Curso de criação de componentes em Delphi  

Unidade 1. Introdução para o curso  <Figura: delphi.gif - 582,0 K> 

------------------------------------------------------------------------

<Figura: Bverde.gif - .325 K>Objetivo do curso. 

Este curso busca oferecer uma aproximação ao desenvolvimento de componentes de um ponto de vista eminentemente prático. Em uma porção de livros de programação em Delphi está o tópico do desenvolvimento de componentes, mas só de um modo rápido e superficial. O modo certo é aquele dos manuais do Delphi que são completamente dedicados a este tópico, mas, a meu modo de ver, sofre de falta de exemplos práticos que vão passo pedagógico a um modo gradual do desenvolvimento de componentes. Também certos tópicos, (p.e o tópico de editores de propriedades) eles permanecem muito no ar. 

Este curso tentará cobrir estas faltas. Este curso busca ser principalmente prático: a teoria da criação de componentes deixará vendo como você precisa. Você começará a ver a teoria básica de criação de componente e imediatamente será aplicado à criação de componentes completamente funcionais. 

Ao longo do curso nós aprenderemos a criar componentes de tipos muito diversos. Nós começaremos com um visual e não-visuais e nós iremos progredindo na criação de componentes gráficos, editores de propriedades, componentes de banco de dados... Também será explicado como criar os arquivos de ajuda para fazer com que nosso componente seja completamente integrado no ambiente de desenvolvimento de Delphi. 

<Figura: Bverde.gif - .325 K>Conhecimento prévio. 

Mas isto não é um curso de programação em Delphi, nem um curso de programação não guiado a objetos. É suposto que o leitor já tenha mais ou menos uma bagagem destes tópicos e tais conceitos como herança, descendente. Como condição prévia para a criação de componentes é altamente aconselhável ter claro os seguintes conceitos: 

•O que a programação é guiada a objetos e seus aspectos fundamentais (fabricantes, destrutivo, classes, herança, sobrecarregam...) •Domínio do ambiente integrado de desenvolvimento de Delphi. •Administrar com facilidade os diferentes componentes standards do Delphi. •Uso do ObjectBrowser para determinar relações entre objetos. 

Se você precisa de alguma explicação nestes tópicos, você pode achar toda a informação necessária nos próprios manuais de Delphi, como também na ajuda on-line. 

<Figura: Bverde.gif - .325 K>Ritmo do curso. 

Em princípio o periodicidade do curso serão semanais ou quinzenais, mas tudo dependerá da sua resposta, os usuários do curso como também de minha prontidão ao longo do tempo (você já sabe, exames e essas coisas). 

<Figura: Bverde.gif - .325 K>Você consulta, dúvidas, sugestões... 

Este curso não estaria completo sem sua colaboração, assim eu espero sua  critica, sugestão, propostas de componentes, exemplos práticos, etc. No seguinte e-mail: revueltaroche@redestb.es 

Se você desenvolveu algum componente que você acredita que pode ser interessante, você esta convidado a compartilhar isto com todos os leitores do curso. 

Não se esqueça: O interesse e a praticidade deste curso também depende de você...

Unidade 2. Um pouco de Teoria  <Figura: delphi.gif - 582,0 K> 

 

------------------------------------------------------------------------

<Figura: Bverde.gif - .325 K>O que são componentes? A biblioteca visual de componentes (VCL). 

Os componentes são a pedra angular da programação em Delphi. Embora a maioria dos componentes representam partes visíveis da interface de usuário, também existem os componentes não-visuais, como por o exemplo o objeto Cronômetro e Banco de dados. 

Um componente, em sua definição mais simples mais é um objeto descendente tipo TComponent. Todos os componentes descendem na forma mais primitiva do TComponent, desde que TComponent provê as características básicas que todo o componente deveria ter: capacidade de ser mostrado na paleta de componentes como também de operar em "Design-time" . 

Os componentes fazem  a programação em Delphi ficar mais facil. Em vez de ter que operar a nível de unidades, o usuário de um componente tem que preencher as propriedades dele e localizar isto na posição querida da forma dele simplesmente. Isso é tudo:O Delphi se encarrega do resto. 

Todos os componentes são parte da hierarquia de Biblioteca de Componente de objetos denominada Visual (VCL). Quando um componente novo é criado, este é derivado a partir de um componente existente (bem como o TComponent ou algum outro especializado) e é somado ao VCL. 

<Figura: Bverde.gif - .325 K>Anatomia de um componente. Propriedades, métodos e eventos. 

  Como já foi mencionado, um componente é um objeto, e como tal, consiste em código e dados. Mas quando se referindo a estes nós não usaremos estas condições, mas nós falaremos bastante de propriedades , métodos e eventos. Ao longo deste curso iremos estudar todos estes aspectos a fundo, mas vamos fazer um pequeno resumo: 

Propriedades 

•As propriedades provêem ao usuário do componente um acesso fácil para o mesmo. Ao mesmo tempo, permite para o programador do componente esconder a estrutura subjacente de dados. Entre as vantagens de usar propriedades para consentir ao componente pode fazer um compromisso: 

•As propriedades estão você pode alterar em design-time. Deste modo o usuário do componente pode inicializar os valores das propriedades sem necessidade de escritura uma linha de código. •As propriedades permitem a validação dos dados para o mesmo tempo de ser introduzidos. Podem ser prevenidos erros deste modo causado por valores inválidos. •Eles nos asseguram que o primeiro valor que colocarmos nas propriedades terão que ser certas e evitarão o erro. 

  

•Eventos •Os eventos são as conexões existentes entre um certo evento e o código escrito pelo programador de componentes. Deste modo por exemplo, antes do clique de evento do mouse poderia se mostrar uma mensagem em tela. Para o código que é executado quando um certo evento o leva a um lugar é denominado Gerenciador de Eventos (o Gerenciador de Eventos) é normalmente escrito pelo usuario do componente. Os eventos mais comuns já são parte dos próprios componentes de Delphi (eventos do mouse, teclado...), mas também é possível definir eventos novos. •Métodos •Os métodos são as funções de E/S de procedimentos que eles são parte do componente. O usuário do componente os usa para obter uma certa ação ou obter um certo valor para o qual você não pode conseguir por meio de uma propriedade. Considerando que eles requerem execução de código, os métodos estão disponíveis só em run-time. 

<Figura: Bverde.gif - .325 K>Controle de acesso para um componente. Declarações: Private, Protected, Public e Pubished. 

O Object Pascal tem quatro níveis de controle de acesso para os campos, propriedades e métodos de um componente. Este controle de acesso permite o programador de componentes em que partes do código especificar as declarações do objeto. Deste modo é definido a interface do componente. É importante planejar bem esta interface, deste modo nossos componentes serão facilmente programáveis e reutilizaveis. 

A menos que o oposto seja especificado, os campos, propriedades e métodos que são somados a um objeto são de tipo published. Todos os níveis de controle de acesso operam a nível de unidades, quer dizer, se uma parte de um objeto é acessível (ou inacessível) em uma parte de uma unidade, é também acessível (ou inacessível) em qualquer outra parte da unidade. 

Logo os tipos de controles de acesso são detalhados: 

<Figura: Bamar.gif - .326 K>Private: escondendo os detalhes de implementação. 

Declarando uma parte de um componente (bem é um campo, propriedade ou método) private (você se priva). Aquela parte do objeto é invisível ao código externo para a unidade no o qual o objeto é declarado. Dentro da unidade que contém a declaração, o código como o que pode consentir àquela parte do objeto se fosse público. 

A utilidade principal das declarações privadas é que eles permitem esconder os detalhes de implementação do componente para o usuário final do mesmo, desde estes eles não podem consentir à parte privada de um objeto. Deste modo você pode mudar o implementação interno do objeto sem afetar ao código que o usuário escreveu. 

<Figura: Bverde.gif - .325 K>Protected: definindo a interface do programador. 

Declarar uma parte de um componente como sendo protegido é o mesmo que declarar  isto em privado. (é escondido ao código externo à unidade). A diferença principal entre declarar uma parte de um objeto protegido ou fazer isto privado são que os descendentes do componente poderão fazer referência àquela parte. 

Este parte é especialmente útil para o criação de componentes. 

  

<Figura: Bverde.gif - .325 K>Public: definindo a interface a tempo de execução. 

As partes inteiras de um objeto que nós declaramos  público, eles poderão ser indexado por qualquer código interno ou externo para a própria unidade. Deste modo, a parte pública identifica a interface a tempo de execução de nosso componente. Os métodos que o usuário do componente deveria chamar os públicos deveria ser declarado, como também as propriedades de read-only, para ser só válido a tempo de execução. 

O público declara as propriedades que não aparecerão no Object Inspector. 

Esta seção é talvez mais importante considerar quando estamos projetando um componente. Quando são projetados componentes, deveriam ser considerados cuidadosamente que métodos e propriedades deveriam ser públicas. Se o código está correto, isto permitirá a alteração as estruturas de dados e métodos internos do componente sem ter que jogar o público da interface que continuará sendo o mesmo para o usuário do componente. 

<Figura: Bverde.gif - .325 K>Published: definindo a interface a design-time: 

Quando declarando parte de um objeto publicado é o mesmo que a pública e também gera a  informação em run-time. 

As propriedades declaradas publicadas aparecem no Object Inspector. Estas partes definem a interface a design-time de nosso componente.

<Figura: Bverde.gif - .325 K>Passos necessários para criar um componente. O expert em componentes. 

As grandes características, os passos necessários para criar um componente novo são o seguinte: 

1.Criar uma unidade para o componente novo. 2.Derivar o componente novo a partir de outro existente, o o qual servirá como base para somar as características do novo componente. 3.Somar as propriedades, eventos e métodos necessários para o componente novo. 4.Registrar o componente, inclusive o bitmaps adicional, arquivos de ajuda, etc. 5.Instalar o componente novo na paleta de componentes. 

De todos os passos mencionados, há um que especialmente é que você tem que se preocupar: a escolha do componente ao qual vai se derivar. Este passo é crucial, com uma boa escolha do ascendente nós podemos economizar muito código e chegar facilmente ao nosso objetivo. 

Como base para a escolha do objeto ascendente, ele serve para fazer para advertência as normas seguintes: 

•TComponent - O ponto de partida para componentes não visuais. •TWinControl - O ponto de partida se é necessário que o componente tenha "handles". •TGraphicControl - Um ponto de partida bom para componentes visuais que não tenham "handles". Esta classe tem os metodos Paint e Canvas. •TCustomControl - O ponto de partida mais comum. Esta classe tem "handle" de janela, eventos e propriedades comuns e, principalmente, os metodos Canvas e Paint. 

Bem, nós já sabemos como determinar o ponto de partida. Deixe nos ver agora como criar a unidade que abrigará o componente. Há duas opções, criar a unidade manualmente ou deixar que o Delphi faça o trabalho sujo. (Use o Expert de componentes). Se nós escolhermos para a primeira opção, teriamos que fazer tudo na mão, mas é aconselhavel utilizar o expert do Delphi. 

Abrir o  expert de componentes no menu Component e New Component... 

Aparecerá uma caixa de diálogo na qual nós devemos preencher os campos: 

•Class Name: Aqui nós deveríamos especificar o nome do componente novo. •Ancestor Type: Nós introduziremos o ascendente aqui a partir do o qual nós derivaremos nosso componente. •Pallete page: Nós indicaremos a página da paleta aqui no o qual nós queremos que o componente novo apareça. 

Uma vez introduzido estes campos, quando pressionamos OK você verá o código de nossa unidade. Se por exemplo nós introduzimos os dados seguintes no expert de componentes: 

Class Name: TMiComponente 

Ancestor Type: TComponent 

Pallete Page: Curso 

Quando clicamos aceitando,O Delphi nos geraria a unidade seguinte, listada abaixo para introduzir as propriedades e métodos de nosso componente: 

    

unit Unit1;

interface

uses

  SysUtils, WinTypes, WinProcs, Messages, Classes, Graphics, Controls,

  Forms, Dialogs;

type

  TMiComponente = class(TComponent)

  private

    { Private declarations }

  protected

    { Protected declarations }

  public

    { Public declarations }

  published

    { Published declarations }

  end;

procedure Register;

implementation

procedure Register;

begin

  RegisterComponents('Curso', [TMiComponente]);

end;

end.

A partir daqui tudo o que consiste em introduzir as propriedades e métodos necessários para a operação de nosso componente. Mas antes de qualquer coisa deve se fazer algumas advrtências em alguns aspectos: 

Na cláusula uses, o Delphi soma por padrão as unidades standards. Se nosso componente não usa alguns deles que nós podemos eliminar isto desta cláusula. Da mesma maneira, se nós usamos algum procedimento ou função localizada em outra unidade, nós deveríamos adicionar esta unidade para a cláusula que uses. 

As declarações das propriedades, campos e métodos que nós vamos definir, os localizará na seção apropriada da interface como corresponde, quer dizer nós os declararemos private, protected, public ou published. 

O Delphi declara e define o registro de procedimento automaticamente para registrar o componente na paleta.

Unidade 3. Um  componente não-visual: TNif  <Figura: delphi.gif - 582,0 K> 

  

------------------------------------------------------------------------

<Figura: nif.gif - .933 K>Chegou o momento de entrar em ação. Nós criaremos nosso primeiro componente: TNif. Mas antes de começar a escrever código isto serve para analisar o propósito do componente e como nós implementaremos isto. 

<Figura: Bverde.gif - .325 K>Objetivo do componente 

O propósito do componente é muito simples: a partir de um dado de entrada, o número de DNI, nosso componente deverá calcular a letra do NIF correspondente. O método de calcular esta letra é por meio de uma operação matemática muito simples. Esta operação é: DNI - ((DNI div 23) * 23). O resultado desta operação será um número entendido entre 1 e 23. Por meio de um quadro nós nomearemos uma certa letra a cada número, a letra que corresponde o NIF pedido. Os detalhes completos desta letra eu tenho no código fonte. 

<Figura: Bverde.gif - .325 K>O projeto do componente 

Baseado em nosso objetivo, está claro que nosso componente será do tipo não-visual e, então, nós derivaremos isto a partir de TComponent que eu como já foi visto; (unidade 2), é a base para a criação de componentes não-visuais. 

A forma de introduzir o número de DNI será por meio de uma propriedade (propriedade DNI) de leitura. O valor desta propriedade será armazenada em um campo (FDNI) do tipoLongInt . Esta propriedade será publicada (published) de forma que ela aparece deste modo no Object Inspector . A leitura e escrita de valores nesta propriedade serão diretamente no campo FDNI, quer dizer, nós deixaremos que o  próprio Object Inspector se encarrega de verificar se o valor introduzido orresponde ao tipo LongInt. 

Nós teremos outra propriedade (NIF) no que a letra calculada será armazenada (de tipo Char). Mas esta propriedade será somente leitura ja que o usuário do componente não poderá inserir a letra manualmente, pois o proprio componente que calcula isto. Esta propriedade deveria ser pública (Public) desde que seja somente leitura. Este aspecto serve para se salientar isto: as propriedades somente-leitura deverião ser declaradas na parte pública. A função que se encarrega de calcular a letra do NIF se chamará GetNIF. 

Sempre serão declarados os campos que armazenam o valor de propriedades na parte privada (Private) desde que nós tenhamos certeza de que todos tenham acesso a ele, mas o usuário do componente, desde que se conscientize pela propriedade e não pelo  campo (isso representa o armazenamento interno da propriedade). 

Relativo aos nomes usados, são continuadas as convenções seguintes: 

•Os tipos que nós definimos T começará com a letra (de tipo). P.e. TNif •Os campos que armazenam os valores de propriedades começam com a carta F seguido pelo nome da propriedade que armazena. Deste modo o campo FDNI está claro que armazena o valor da propriedade DNI. •Os nomes dos métodos de leitura e dos valores de uma propriedade, eles serão denominados pelo prefixo Get (para ler) ou Set (para escrever) seguido pelo nome da propriedade. P.e. o método GetNIF. 

Uma vez o método de empregado de desígnio de acordo é hora de começo para digitar. 

<Figura: Bverde.gif - .325 K>Codifique fonte do componente 

    

unit Nif;          { (c)1996 by Luis Roche }

interface

uses

  Classes;

type

  TNif = class(TComponent)   

  private

    FDNI : LongInt;          

    function GetNIF : char;  

  protected

  public

    property NIF: char read GetNIF;    

  published

     property DNI: LongInt read FDNI write FDNI;  

  end;

procedure Register;       

implementation

function TNIF.GetNIF : char;      

Var aux1 : integer;

Const letras : string = 'TRWAGMYFPDXBNJZSQVHLCKE';

begin

  aux1:=FDNI - ((FDNI div 23) * 23);

  result:=letras[aux1+1];

end;

procedure register;           

begin

  registercomponents('curso', [tnif]);

end;

end.

<Figura: Bverde.gif - .325 K>Comentários para o código fonte 

O código fonte tem que ser criado através dos seguintes passos: 

•Nós usamos o Expert de componentes de forma que nós geramos a unidade na qual nós escrevemos nosso componente. Na Caixa de diálogo que o expert nos mostrou, nós introduzimos TNif como Nome da Classe Name, TComponent como Tipo de Ascendente e Curso como Nome da Paleta. Quando clicamos OK, o expert de componentes cria todo o esqueleto básico de nosso componente, inclusive o Registro de procedimento. 

De todas as unidades que aparecem na cláusula uses nós só deixaremos a unidade Classes, pois nós não usaremos qualquer procedimento dos outros. 

Nós declaramos o campo FDNI (longInt) na seção privada. Neste campo, como já foi dito, nós armazenaremos o DNI. Na seção published nós escrevemos a linha seguinte: 

propriedade DNI: LongInt read FDNI write FDNI 

Deste modo nós declaramos a propriedade DNI e nós especificamos que a leitura e escrita de valores no mesmo são feitos diretamente com o campo FDNI que usa o Object Inspector. 

Nós declaramos a propriedade NIF somente-leitura na parte publica. Ele serve para se lembrar que as propriedades de leitura só deveriam ir declaradas na parte pública e não no published. Nós especificamos isso para ler o valor da propriedade: nós usaremos a função GetNIF, o qual nós declaramos na seção privada. 

Nós escrevemos a função que calcula a letra do NIF na parte de implementação da unidade. 

Nós mantemos a unidade com o nome nif.pas. Veste que todos os componentes que nós vamos criar durante o curso separados em diretório (delphi\componen)

<Figura: Bverde.gif - .325 K>Criando um bitmap para o componente 

Cada componente precisa de um bitmap para representar ao componente na paleta de componentes. Se nenhum for especificado, o Delphi usará um como padrão. 

O bitmap não é incluído no código fonte do componente, mas é incluído separado em um arquivo com a extensão .DCR (Dynamic Component Resource). Este arquivo pode ser criado com o próprio editor de imagens que vem com o Delphi. 

O nome do arquivo que .DCR deveria coincidir com o nome da unidade do seu componente. O nome do bitmap de imagem (em letra maiúscula) devera coincidir com o nome do componente. Os dois arquivos (o da unidade * .pas e que do bitmap * .dcr) devem residir no mesmo diretório. 

Em nosso componente, se nós salvamos a unidade com o nome nif.pas  o arquivo de recursos se chamará nif.dcr. Dentro deste arquivo estarão o bitmap para os quais nós colocaremos o nome TNIF. O bitmap que nós criamos: <Figura: nif.gif - .933 K> O tamanho do bitmap devera ser de 24x24 pixels. 

Como último detalhe, se você quer usar este mesmo bitmap, você pode usar um programa de tratamento de imagens para recortar isto e colar no editor de imagens do Delphi. 

<Figura: Bverde.gif - .325 K>TNif instalando na paleta de componentes. 

Para instalar nosso componente na paleta de componentes é muito simples. Siga os passos: 

•Escolher o menu Componentes de Options|Install . •Clicar no botão Add, selecionar o local que contém o código fonte do componente. Em nosso caso, nif.pas. Uma vez selecionado a unidade, clicar OK. •O Delphi compilará a unidade e reconstruirá o arquivo COMPLIB.DCL. Se ele gerar erros durante a compilação, nós seremos informados disto. •Se a compilação tem êxito, nosso componente fara parte da paleta de componentes.

<Figura: delphi.gif - 582,0 K> Curso de criação de componentes em Delphi   

Unidade 4. Adicionando recursos a um componente existente: TBlinkLabel  <Figura: delphi.gif - 582,0 K> 

  

------------------------------------------------------------------------

<Figura: blinklbl.gif - .966 K>Nesta unidade nós aprenderemos como já adicionar recursos e possibilidades para um componente Delphi existente. Nós aprenderemos a usar objetos (componentes) em nosso próprio componente, o que eles são e como se declaram os constructors e os destructors. 

Tudo isso,claro é, baseado em um exemplo concreto: TBlinkLabel. 

<Figura: Bverde.gif - .325 K>Objetivo do componente 

Nosso propósito é criar um componente idêntico ao o tipo TLabel , mas com uma capacidade a mais: o de piscar. Quer dizer, nosso componente deveria nos mostrar a mensagem introduzida na de propriedade que pisca com uma certa freqüência. Também, se o valor introduzido freqüência é nulo, a mensagem deverá permanecer fixa na tela (emulando o que acontece a um Tlabel normal). 

<Figura: Bverde.gif - .325 K>Usando um componente existente em nosso próprio componente. Cronômetros. 

Como nós há pouco dissemos, nosso componente deveria piscar com uma certa freqüência. Porque nós precisamos ser possíveis medir o tempo que devera piscar a mensagem. Afortunadamente, o Delphi nos provê um objeto que molda perfeitamente a nossas exigências de medição de tempo: TTimer. 

Embora não é este o lugar de explicar as características do Cronômetro de objeto (a informação básica deste componente pode estar na própria ajuda on-line do Delphi), ele serve para parar um momento na propriedade do cronômetro que nós usaremos: o intervalo da propriedade. Esta propriedade nos permite especificar o intervalo de tempo que nós queremos medir (em milisegundos). Quando o valor estabelecido é alcançado o método que esta especificado no evento OnTimer é executado. 

O método que nós especificaremos para o evento OnTimer fara com que a mensagem apareca e desapareça. 

A teoria é muito simples, mas... como nós adicionamos o objeto cronômetro para nosso componente? Em design-time é simples. É só selecionar o cronômetro na paleta de componentes e colocá-lo em nossa forma. 

Mas quando estamos escrevendo um componente nós não podemos usar este sistema, se nós não vamos isto à mão. Os passos gerais que devem ser seguidos: 

•Adicionar a unidade que o componente é definido na claúsula uses de nossa unidade. Como o objeto cronômetro é declarado na unidade ExtCtrls , é esta unidade que o que nós deveríamos somar ao claúsula uses. •Adicionar o método necessário para construir o objeto (em nosso caso, o cronômetro). •Escrever o código necessário para destruir o objeto no método destructor de nosso componente. 

Na seção seguinte nós aprofundaremos nestes dois passos. 

<Figura: Bverde.gif - .325 K>Construtores e Destruidores. 

<Figura: Bamar.gif - .326 K>Construtores 

Os objetos que nós declaramos em um componente não existem em memória até que objeto é criado (também é dito que uma instância do objeto é criada) para uma chamada para o método de construção do objeto. 

Um Construtor não é mais que um método que provê memória para o objeto e aponta (por meio de um ponteiro) para ele. Chamando ao método create, o construtor nomeia a instância do objeto para uma variável. 

Ele serve para fazer advertência que todos os componentes herdam um método denominado create que se encarrega de criar o componente em memória. De um modo adicional, no método create os valores de certos campos do componente podem ser inicializados (comumente lhes nomear um valor por padrão) 

No método create do componente nós podemos criar objetos adicionais que ele precisa para nosso componente. Em nosso caso, no método create de TBlinkLbl nós criaremos o cronômetro. 

<Figura: Bamar.gif - .326 K>Destruidores 

Quando nós terminamos de usar um objeto, nós deveríamos destruir isto, quer dizer, liberar a memória que ocupou o objeto. Esta operação é feita atraves do método destroy que todos os componentes herdam de TComponent. 

Se nós criamos algum objeto adicional, nós deveríamos destruir também escrevendo isto noo  método destroy.

Delphi cria e destrói nosso componente automaticamente quando é necessário, como ja foi dito, os métodos create e destroy são herdados do TComponent. 

Mas se nós queremos usar algum objeto em nosso componente, o Delphi não cria e destroi automaticamente, nós devemos fazer isto manualmente. Este processo é feito para adicionar o código necessário para os métodos  create e destroy. É importante dizer que nós escreveremos o mesmo código novamente no componente, e podem sobrecarregar o código novo. 

De um modo geral isto é feito do seguinte modo: (um exemplo mais concreto; isto tem na próprio código fonte do componente) 

•Na parte pública de nosso componente nós declaramos os métodos o create e destroy seguido pela palavra chave overrride: •      public

                       constructor create(AOwner : TComponent); override;

                       destructor   destroy; override;

•Escrever o código adicional do construtor e do destruidor na seção de implementação. •    

                constructor TComponent.Create(AOwner : TComponent);

                     begin

                          inherited Create(AOwner);

                          ...

                     end;

                     destructor TComponent.Destroy;

                     begin

                          ...

                          inherited destroy;

                     end;

A coisa mais importante para se lembrar de é que quando sobrecarregamos um construtor, a primeira coisa que deveria ser feita é uma ligação ao construtor original (linha herdada o create). Próximo a pessoa pode somar o código necessário para criar o objeto. 

Da mesma maneira, quando sobrecarregamos um destruidor, os objetos criados antes deveriam ser liberados e a última linha deveria ser uma chamada ao destruidor original.

<Figura: Bverde.gif - .325 K>Implemementando o piscador. 

Em design-time do componente nós deixamos ainda um passo importante: como fazer as piscadas de mensagem. Nós dissemos que quando o evento acontece OnTimer o método será executado e mostrará e ocultara a mensagem dando uma sensação de "pisca-pisca" 

A solução é muito simples, desde que nosso componente, descendendo de um TLabel uma propriedade que determina se o componente deveria ser visível ou não. Esta propriedade é denominada visible. 

Deste modo, nosso método pisca alternará o valor da propriedade booleana visible quando o evento OnTimer acontece. São mostrados os detalhes concretos de implementação no e código fonte. 

<Figura: Bverde.gif - .325 K>Outros detalhes na criação do componente. Valores padrão. 

Desta forma,  nós concluimos dois aspectos na criação do componente: 

•Nós comentamos que se o valor introduzido na propriedade  freqüência for nulo, a mensagem deveria permanecer fixa em tela. Isto é alcançado ativando e incapacitando o componente quando é introduzido o valor da velocidade de propriedade no Object Inspector (ver o procedimento SetVelocidad no código fonte). •

Deste modo, se é introduzido 0 como valor da velocidade, o cronômetro é simplesmente inválido e ele usa o valor da propriedade visível para True. Se o valor introduzido é diferente de zero, o cronômetro é habilitado e, por conseguinte, o pisca - pisca. •Um último detalhe é dar um valor padrão para a propriedade que define a velocidade. Isto é adquirido em dois passos: primeiramente na declaração da propriedade velocidade a palavra chave Default seguida pelo valor é somada por padrão (400 em nosso caso). Logo que o construtor é inicializado o campo associado com a propriedade (FVelocidad:=400). Os dois passos são necessários e não são possíveis não obviar nenhum de ambos. •

Este detalhe de nomear valores padrão é mais importante para evitarmos erros de inicialização e temos certeza de que a propriedade sempre terá um valor válido. O Delphi só manterá o valor da propriedade se ele for diferente ao valor padrão. 

<Figura: Bverde.gif - .325 K>O Código Fonte do componente 

    

unit Blinklbl;      { (c)1996 by Luis Roche }

interface

uses

  Classes, StdCtrls, ExtCtrls;    

type

  TBlinkLabel = class(TLabel)

  private

    FVelocidad : integer;

    FTimer : TTimer;            

    procedure SetVelocidad(valor : integer);  

  protected

    procedure parpadea(Sender : TObject);

  public

    constructor Create(AOwner : TComponent); override;    {Constructor}

    destructor Destroy; override;                         {Destructor}

  published

    property Velocidad : integer read FVelocidad write SetVelocidad default 400;

  end;

procedure Register;

implementation

constructor TBlinkLabel.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);       {Chama o constructor original (herdado)}

  FTimer := TTimer.Create(Self);  {Criamos o timer}

  FVelocidad := 400;              {Freqüência (velocidade) padrão}

  FTimer.Enabled:=True;           {Ativamos o timer}

  FTimer.OnTimer:=parpadea;       {Pegamos o método parpadea (Pisca-Pisca)}

  FTimer.Interval:=FVelocidad;    {Pegamos o intervalo do timer = freqüência parpadeo}

end;

destructor TBlinkLabel.Destroy;

begin

  FTimer.Free;          {Liberamos o timer}

  inherited destroy;    {Chamamos o destructor original (herdado)}

end;

procedure TBlinkLabel.SetVelocidad (valor : integer);

begin

  If FVelocidad <> valor then        {Só se o valor introduzido é diferente do padrão}

  begin

    if valor < 0 then FVelocidad:=0;

    FVelocidad:=Valor;               {Aceita a velocidade}

    if FVelocidad=0 then FTimer.Enabled:=False else FTimer.Enabled:=True;

    {Se Fvelocidad=0 a menssagem deve estar sempre visible}

    FTimer.Interval:=FVelocidad;

    Visible:=True;

  end;

end;

procedure TBlinkLabel.parpadea(Sender : TObject);

begin

  if FTimer.Enabled then Visible := not(Visible);  {Alternativamente mostra e oculta a menssagem se o timer esta activado}

end;

procedure Register;        {Registro do componente}

begin

  RegisterComponents('Curso', [TBlinkLabel]);

end;

end.

Curso de criação de componentes em Delphi

  Unidade 5. O objeto Canvas e o método Paint: TGradiente  <Figura: delphi.gif - 582,0 K> 

------------------------------------------------------------------------

<Figura: blinklbl.gif - .966 K>Nesta unidade nós criaremos um componente gráfico do tipo que é usado como fundo na instalação de aplicações: TGradient. Mas nosso componente irá mais adiante ele terá possibilidades novas para o gráfico standard. Nós aprenderemos a utilizar o canvas que prepara todos os componentes gráficos. Nós estudaremos o método paint, como usar isto em nossos componentes. 

<Figura: Bverde.gif - .325 K>Objetivo do componente 

Nosso propósito é criar um componente no qual nós puxaremos um gradient semelhante de cores para qual é mostrado na maioria dos programas de instalação de aplicações (e em outro tipo de programas). Entre as características adicionais que nosso componente deve ter é necessário  destaquar que o gradiente de cores não concluirá na cor preta, mas  poderá começar e terminar em duas cores quaisquer. Também, o gradiente poderá estar horizontal ou vertical. Ultimamente, o tamanho de nosso desenho (quer dizer, do canvas do componente) será variável, deste modo nós podemos combinar dois ou mais componente para criar efeitos de cores espetaculares. 

Nosso componente derivará então de TGraphicControl do que nós precisamos que tem Canvas, mas não é necessário que tenha manipulador (o handler). Esta escolha faz isto baseado no que foi visto na unidade 2. 

<Figura: Bverde.gif - .325 K>A base da programação gráfica em Delphi: o  Canvas do objeto. 

Tradicionalmente, a programação gráfica em janelas foi levada  por meio do uso da interface gráfica de dispositivo (GDI). Esta interface é uma ferramenta poderosa que permite o desenho de gráficos por meio do uso de pincéis, escovas, retângulos, elipses, etc. 

Mas o GDI tem uma inconveniência: a programação dele é muito laboriosa. Quando eles fazem uso da  função GDI, é necessária uma handle, como também criar e destruir as ferramentas de desenho diversos (recursos) isso é usado. Por último , quando concluindo, o contexto de dispositivo deve ser restabelecido a seu estado original antes de destruir-lo. 

Delphi encapsula o GDI e faz que nós não tenhamos que nos preocupar sobre contextos de dispositivo nenhum e se nós liberamos ou não os recursos usados. Deste modo nós podemos nos concentrar em  nosso objetivo principal: puxar os gráficos. De todos os modos, se nós queremos isto, nós podemos continuar usando as funções GDI se nos interessa. Nós temos um grande leque de possibilidades, Delphi ou GDI, podemos usá-los do modo que nos convenha. 

Como foi mencionado no parágrafo anterior, o Delphi nos provê uma interface gráfica completo. O objeto principal desta interface é o objeto Canvas . O objeto Canvas se encarrega de ter um contexto de dispositivo válido, como também de liberar -se quando nós não usarmos mais. Também, o objeto Canvas (ou simplesmente Canvas) tem propriedades diversas que representam o lápis atual (pen), escova (brush) e fonte (font). 

O canvas administra todos estes recursos para nós, razão por que nós temos bastante o que informar à aquela classe de caneta e ele se encarrega do resto. Também, deixando o Delphi fazer o serviço pesado de criar e liberar os recursos gráficos, em muitos casos um aumento de velocidade acontecerá se nós mesmos administrarmos estes recursos. 

A tela de objeto encapsula a programação gráfica a três níveis de profundidade que é o seguinte: 

  

Nível Operação Ferramentas Alto Desenho de linhas e formas  

Visualização e modificação de texto  

Recursos de áreas Métodos MoveTo, LineTo, Retângulo, Elipse,  

Métodos TextOut, TextHeight, TextWidth, TextRect,  

Métodos FillRect, FloodFill, Médio Personalizar texto e gráficos  

Pixels de manipulação  

Copia e união de imagens Propriedades Pen, Brush e Font  

Propriedade Pixels  

Draw, StretchDraw, BrushCopy, CopyRect, CopyMode, Baixo Chamadas para funções GDI Handle de propriedade 

  

Agora nós não vamos explicar tudo e cada ferramenta disponível para a criação de gráficos, mas nós faremos isto como nas proximas unidades. De todos os modos, uma descrição detalhada de cada um deles está na ajuda on-line do delphi 

Bem, nós já sabemos que a tela de objeto nos provê uma interface gráfica muito poderosa de criação gráfica. Mas agora as perguntas seguintes podem ser esboçadas: Onde o objeto Canvas reside?, todos os componentes têm isto?, e se não é deste modo, como os componentes incorporam isto? 

A resposta para estas perguntas é simples: Todos os objetos derivados de TGraphicComponent possuem Canvas. Em relação a outros componentes, depende. No caso de dúvida, a coisa mais simples é consultar o Object Browser e verificar se o componente em questão possui canvas ou não (se é declarado em um modo protegido ou público). Deste modo por exemplo, o componente Tlabel, Timage e TPanel possuem isto, enquanto tais componentes como TButton ou TRadioButton não. Este aspecto determinará o ascendente a ser escolhido quando nós queremos desenvolver um componente que devera ter Canvas. 

A forma de consentir o canvas de um objeto é muito simples. Vamos supor que nós temos um componente (TGradient) derivou da classe TGraphicControl. E vamos supor que nós queremos tirar no canvas uma linha do ponto (0,0) até o ponto (20,20). Para esta tarefa nós deveríamos escrever a  seguinte linha: 

TGradient.Canvas.MoveTo(0,0);    

TGradient.Canvas.LineTo(20,20);

Vamos ver agora as propriedades do Canvasa que nós usaremos no desenvolvimento de nosso componente. Basicamente TGradient deveria puxar uma série de retângulos coloridos com uma certa cor. Para isso nós usaremos o método FillRect. O método FillRect recebe como parâmetro um objeto do tipo TRect com as coordenadas superiores esquerdas e direito inferior para pintar. Deste modo, o código que nós usaremos em nosso componente será semelhante ao seguinteo (nós ainda não nos definimos a cor de preenchimento) 

Canvas.FillRect(TRect);

Para escolher a cor de preenchimento nós usaremos a  propriedade brush do canvas. A propriedade brush determina a cor  padrão que o canvas usa para preencher formas gráficas e fundos. Em nosso caso, nós usaremos a propriedade color do brush para escolher a cor que preencherá o canvas. 

Nós também precisaremos usar a  propriedade pen do canvas. O objeto pen determina a classe de lápis que nós usaremos no canvas para puxar linhas e pontos. Em nosso caso, nós usaremos a propriedade estilo do pen que determina o tipo de linha para puxar (em nosso componente psSolid) e o modo de propriedade que determina o modo com que o lápis utilizará o canvas (em nosso componente pmCopy ) 

<Figura: Bverde.gif - .325 K>Criando o gradiente de cores. 

No Delphi, uma cor é representada através de 4 bytes hexadecimais. O byte mais alto é usado para determinar o ajuste que o Windows faz da paleta e não veremos isto com mais detalhe. Os outros três bytes (os três bytes mais baixos) eles representam a quantidade de vermelho, verde e azul que forma a cor. Nosso componente devera calcular um gradiente de cores de um inicial (FColorDesde no código fonte) até outro fim (FColorHasta). O melhor modo para calcular este gradiente é decompondo as cores RGB (Vermelho, Verde e Azul) em seu componente. Deste modo, nós saberemos a quantidade de cada uma destas três cores básicas que irão formar uma certa cor. Por exemplo a cor vermelha pura é representada por 255,0,0 (255 de vermelho, 0 de verde e azul), e um tom cinza tem os três valores de vermelho, verde e mesmo azul (p.e. 150,150,150). 

O decomposição de uma cor em seus três componentes básicos depende de três funções: GetRValue, GetGValue e GetBValue que pertencem ao API de Windows. Estas funções levam como parâmetro a cor da qual nós queremos obter o decomposição e eles devolvem a " quantidade " respectivamente de vermelho, verde e azul que compõe a cor. 

No código fonte, uma vez decompostas as cores inicial e final nas variáveis RGBDesde[0 ..2] e RGBHasta[0 ..2] (0 para cor vermelha, 1 para verde e 2 para azul), o processo de calculo do gradiente é o seguinte: 

1.Calcular a diferença em valor absoluto entre RGBDesde e RGBHasta para cada uma das cores básicas e manter isto na variável RGBDif[0 ..2]. Adicionalmente, no fator variável nós manteremos  +1 se RGBHasta for maior que RGBDesde (gradiente superior) e  -1 caso contrário (gradiente descendente). Este processo é feito para cada uma das três cores básicas. 2.Entre  0 a 255 (nosso gradiente consistirá em 256 cores), nós calculamos a cor que corresponde a cada requadro a puxar (com o método FillRect) por meio da expressão: 3.Vermelho:=RGBDesde[0]+factor[0]*MulDiv(contador,RGBDif[0],255);

(de um modo semelhante para o verde e o azul) 

Nota: MulDiv é uma função do Api de Windows que multiplica os primeiros dois valores passados como parâmetros e o resultado é dividido pelo terceiro parâmetro que devolve o valor desta divisão em forma de 16 bits arredondados. 

4.Nós colocamos a cor calculada na propriedade color do objeto brush: 5.Canvas.Brush.Color:=RGB(Vermelho,Verde,Azul);

Como é fácil supor, a função RGB leva três valores de vermelho, verde e azul, e forma a cor que corresponde a estas cores básicas. 

6.O requadro é tirado: 7.Canvas.FillRect(Banda);

Banda é um tipo de variável TRect que mantém as coordenadas do requadro para puxar.

Ultimo detalhe: como foi mencionado, nosso gradiente consistirá em 256 cores. Porém, dependendo do modo gráfico que nós configurarmos o Windows e da existência de outros objetos com as próprias cores , o Windows ajustará as cores disponíveis de forma que o resultado final sera a mais proxima possível. 

<Figura: Bverde.gif - .325 K>Quando desenhar no Canvas. A mensagem WM_PAINT e o método Paint. 

Quando nós queremos utilizar o canvas, nós devemos colocar o código correspondente dentro do método paint que pertence ao objeto TGraphicControl (o objeto TCustomControl também tem este método). Isto é deste modo porque quando nosso componente deveria ser puxado (p.e. o gradiente de cores), bem porque é a primeira vez em que ou em resposta para uma aplicação de Windows, o Windows enviará  uma mensagem do tipo WM_PAINT para nosso componente. O componente, de um modo automático (herança), faz uma ligação ao método paint para desenhar o componente (em nosso caso, o gradiente). 

Em geral nós não teremos a necessidade de investigar mais a mensagem WM_PAINT. Como nós há pouco explicamos, a única coisa que nos interessa dele (pelo menos para nosso componente), é que quando esta mensagem é recebida, o Delphi executará o método paint associado ao componente. E é dentro deste método que nós devemos adicionar o código necessário para puxar o gradiente. 

Então, os passos  que nós vimos anteriormente que são necessários puxar o gradiente, eles serão localizados dentro do método TGradiente.Paint que será declarado como nula.O método paint declarará isto no tipo protegido, desde que não se possa usá-lo fora de nosso componente. 

<Figura: Bverde.gif - .325 K>Outros detalhes no desenvolvimento do componente. Publicando propriedades herdadas. 

<Figura: Bamar.gif - .326 K>Um dos detalhes que nós veremos agora é o alinhamento do componente. O objeto TGraphicControl possui a propriedade Align (então nosso componente herdará isto), mas será declarado como público. Para nós é interessante declarar isto como published de forma que ela apareca no Object Inspector em design-time. Para isto nós devemos redeclarar a propriedade align na seção published: 

published    

  property align

É importante fazer uma advertência em dois aspectos: o primeiro é que uma redeclaração pode serr só menos restritivo o acesso para uma propriedade (p.e protected a public), mas não mais restritivo (public a protected). 

O segundo aspecto é ao redeclarar, não é necessário especificar o tipo da propriedade, apenas indicar seu nome. O Que  nós podemos fazer no momento do redeclaração é definir um valor novo por padrão para esta propriedade. 

<Figura: Bamar.gif - .326 K>O resto dos detalhes do codigo fonte não precisa de mais detalhes. São declarados as propriedades Direção, ColorDesde e ColorHasta e são escritos os valores correspondentes... 

<Figura: Bverde.gif - .325 K>O Código Fonte do Componente

    

unit Gradient;

interface

uses

  Classes, Controls, Graphics, WinTypes, WinProcs;

type

  TDireccion = (dHorizontal, dVertical);  {Tipo de dirección del gradiente}

  TGradiente = class(TGraphicControl)

  private

    FDireccion : TDireccion;   {Dirección del gradiente}

    FColorDesde, FColorHasta : TColor;     {Color del gradiente}

    procedure SetDireccion(valor : TDireccion);

    procedure SetColorDesde(valor : TColor);

    procedure SetColorHasta(valor : TColor);

  protected

    procedure Paint; override;

  public

    constructor Create(AOwner : TComponent); override;

  published

    property Direccion : TDireccion read FDireccion write SetDireccion default dHorizontal;

    property ColorDesde : TColor read FColorDesde write SetColorDesde default clBlue;

    property ColorHasta : TColor read FColorHasta write SetColorHasta default clBlack;

    property Align;     {Redeclaração da propriedade como publicada}

  end;

procedure Register;

implementation

constructor TGradiente.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);       {Sempre a primeira coisa a fazer}

  FDireccion:=dHorizontal;        {Valores padrões}

  FColorDesde:=clBlue;

  FColorHasta:=clBlack;

  Width:=100;

  Height:=100;

end;

procedure TGradiente.SetDireccion(Valor : TDireccion);

begin

  if FDireccion <> valor then

  begin

    FDireccion := Valor;

    Repaint;              {Força o desenho do gradiente}

  end;

end;

procedure TGradiente.SetColorDesde(Valor : TColor);

begin

  if FColorDesde <> Valor then

  begin

    FColorDesde := Valor;

    Repaint;            

  end;

end;

procedure TGradiente.SetColorHasta(Valor : TColor);

begin

  if FColorHasta <> Valor then

  begin

    FColorHasta := Valor;

    Repaint;              

  end;

end;

procedure TGradiente.Paint;

var

  RGBDesde, RGBHasta, RGBDif : array[0..2] of byte;  {Cores inicial e final - diferença de cores}

  contador, Vermelho, Verde, Azul : integer;

  Banda : TRect;                                     {Coordenadas do requadro a pintar}

  Factor : array[0..2] of shortint;                  {+1 se gradiente é crescente e -1 caso decrescente}

begin

  RGBDesde[0]:=GetRValue(ColorToRGB(FColorDesde));

  RGBDesde[1]:=GetGValue(ColorToRGB(FColorDesde));

  RGBDesde[2]:=GetBValue(ColorToRGB(FColorDesde));

  RGBHasta[0]:=GetRValue(ColorToRGB(FColorHasta));

  RGBHasta[1]:=GetGValue(ColorToRGB(FColorHasta));

  RGBHasta[2]:=GetBValue(ColorToRGB(FColorHasta));

  for contador:=0 to 2 do         {Calculo de RGBDif[] e factor[]}

  begin

    RGBDif[contador]:=Abs(RGBHasta[contador]-RGBDesde[contador]);

    If RGBHasta[contador]>RGBDesde[contador] then factor[contador]:=1 else factor[contador]:=-1;

  end;

  Canvas.Pen.Style:=psSolid;           

  Canvas.Pen.Mode:=pmCopy;              

  if FDireccion = dHorizontal then      

  begin

    Banda.Left:=0;                     

    Banda.Right:=Width;

    for contador:=0 to 255 do

    begin

      Banda.Top:=MulDiv(contador,height,256);

      Banda.Bottom:=MulDIv(contador+1,height,256);

      Vermelho:=RGBDesde[0]+factor[0]*MulDiv(contador,RGBDif[0],255);

      Verde:=RGBDesde[1]+factor[1]*MulDiv(contador,RGBDif[1],255);

      Azul:=RGBDesde[2]+factor[2]*MulDiv(contador,RGBDif[2],255);

      Canvas.Brush.Color:=RGB(Vermelho,Verde,Azul);

      Canvas.FillRect(Banda);           

    end;

  end;

  if FDireccion = dVertical then         

  begin

    Banda.Top:=0;

    Banda.Bottom:=Height;

    for contador:=0 to 255 do

    begin

      Banda.Left:=MulDiv(contador,width,256);

      Banda.Right:=MulDIv(contador+1,width,256);

      Vermelho:=RGBDesde[0]+factor[0]*MulDiv(contador,RGBDif[0],255);

      Verde:=RGBDesde[1]+factor[1]*MulDiv(contador,RGBDif[1],255);

      Azul:=RGBDesde[2]+factor[2]*MulDiv(contador,RGBDif[2],255);

      Canvas.Brush.Color:=RGB(Vermelho,Verde,Azul);

      Canvas.FillRect(Banda);

    end;

  end;

end;

procedure Register;

begin

  RegisterComponents('Curso', [TGradiente]);

end;

end.

Curso de criação de componentes em Delphi  

Unidade 6. TMultiGrid: cores, alinhamento e linhas múltiplas em um TStringGrid.  

<Figura: delphi.gif - 582,0 K> 

------------------------------------------------------------------------

<Figura: blinklbl.gif - .966 K>Nesta unidade nós criaremos  um componente melhorado do  tipo StringGrid. Nós aprenderemos a criar eventos que nos permitirão dotar de possibilidades novas e potentes a nossos componentes. Também, nós estudaremos o método que OnDrawCell e  já nos aprofundaremos nos tratados de tópicos relacionando unidades prévias para os objetos Canvas e Brush. 

<Figura: Bverde.gif - .325 K>Objetivo do componente 

Como nós há pouco mencionamos, nosso propósito é criar um componente do tipo StringGrid mas com funcionalidades novas. De forma que a primeira coisa que nós deveremos saber é o que o componente TStrinGrid padrão nos permite fazer e o que não nos permite. Assim se você ainda não conhece este objeto, nós faremos uma pequena pausa  para vc parar e olhar na ajuda online do Delphi.  Já está? Bem, porque agora as características novas que nós implementaremos a nosso componente: a cor e o alinhamento das celulas ao nível que nós queremos (coluna, linhas e até mesmo celulas individuais, incluindo alinhamento vertical) como também uma propriedade nova denominada multilinha que nos permitirá mostrar mais que uma linha de texto em cada celula do grid . 

A figura que abaixo mostra um exemplo do componente em operação: 

<Figura: mgej1.gif - 4679,0 K>

<Figura: Bverde.gif - .325 K>Implementando o alinhamento de celulas. Criando nossos próprios eventos. 

Nós começaremos com a propriedade de alinhamento. A primeira coisa que nós deveremos decidir é como nós implantaremos isto. Fundamentalmente, há três possibilidades: 

•Definindo o alinhamento de um modo global. Se você optar por esta  implementação, você apenas teria que definir uma propriedade (denominada, logicamente, Alignment) isso controlaria o alinhamento de todas as celulas do grid. Isto seria muito simples, mas muito não muito prático, desde que o que queremos ter são: as celulas de cabeçalho centralizadas, textos alinhados à esquerda, ou  números à direita, etc. 

Uma solução um pouco melhor: definir o alinhamento a nível de colunas. Deste modo cada coluna pode ser alinhada com independênte da outra, mas mesmo assim todas as linhas desta coluna terá o mesmo alinhamento (cabeçalho e dados). Se vc optar por esta solução você terá um problema na sua implementação pois deveríamos criar um editor de propriedades, e nós ainda não sabemos como fazer isto (paciência, que será visto nas próximas unidades). Este é o caso do editor de colunas que Delphi 2 incorpora. 

A terceira possibilidade nos oferece um controle total: definição do alinhamento de cada celula a nível individual. Deste modo cada celula terá seu próprio alinhamento com independente do outro.O problema é que isto requer um pouco mais que esforço por parte do usuário do componente.

Como você vê , cada uma das três opções acima tem suas vantagens e desvantagens. Por isso em nosso componente nós implementaremos uma combinação do primeiro e terceiro método. Deste modo, nós definiremos uma propriedade de Alinhamento específica para o alinhamento individual por celula e outro padrao para alinhamento total da grade. 

A implementação do alinhamento horizontal a nível global não tem nenhum mistério: Basta definir o propriedade Alignment do tipo TAlignment (já incluida no Delphi). O campo que ela manterá o valor desta propriedade terá o nome: FAlignment. Para escrevermos o valor na propriedade, usaremos o método SetAlignment, e a leitura sera feita diretamente no campo FAlignment. Deste modo nós definimos a interface da propriedade Alignment perfeitamente. Nós definiremos as propriedades para o alinhamento individual por celula quando estudarmos o evento OnDrawCell. 

O alinhamento vertical é desenvolvido de um modo semelhante. O único aspecto diferencial é que o Delphi não incorpora a propriedade do tipo TVerticalAlignment, de forma que nós deveremos cria-lo: 

    

   TVerticalAlignment = (vaTop, vaCenter, vaBottom);

Nós veremos agora como implementar a interface do alinhamento das celulas a nível individual. Para isso nós criaremos um novo evento, que ao ser ativado saberemos o alinhamento de uma celula qualquer. 

Como já você viu na unidade 2, um evento é um mecanismo que víncula uma ação para certo código. Mais concretamente, um evento é um ponteiro ao método. 

A implementação do evento é feito por meio de propriedades, quer dizer, os eventos são propriedades. Ao contrário das propriedades padrões, os eventos não usam métodos para implementar as partes lidas e escritas. No lugar delas, as propriedades de evento utilizam um campo privado. 

Mas já temos bastante de teoria, vamos trabalhar!. Como já foi mencionado, nós criaremos um evento novo que deverá ser ativado toda vez que precisarmos obter o valor do alinhamento de uma celula específica. A primeira coisa que nós devemos fazer então, é definir nosso tipo de evento. Isto se faz  por meio seguinte linha: 

    

  TMultiGridAlignmentEvent=procedure(Sender:TObject; ARow,ACol:LongInt;    

    var HorAlignment: TAlignment; var VerAlignment: TVerticalAlignment) of object;

Os parâmetros do evento TMultiGridAlignmentEvent: 

•Sender: Identifica o objeto que faz a ligação. •ARow e ACol: Identificam as coordenadas de Linha(ARow) e Coluna (ACol) para o alinhamento. •

var HorAlignment e var VerAlignment: Parâmetros para alinhamento horizontal (HorAlignment) e vertical (VerAlignment). 

  

Nós já definimos o tipo de evento. Agora nós deveremos criar um campo que contenha o estado da propriedade OnGetCellAlignment. Isto devera ser feito na parte Private: 

    

  private    

     ...    

     FOnGetCellAlignment: TMultiGridAlignmentEvent;

Finalmente, nós definiremos a propriedade na parte Published: 

    

    property OnGetCellAlignment: TMultiGridAlignmentEvent read FOnGetCellAlignment write FOnGetCellAlignment;

Nós só ativaremos o evento quando precisarmos, embora um pouco mais tarde faremos com mais detalhes em nosso componente, a linha abaixo nos mostra um  mecanismo geral: 

    

  if Assigned(FOnGetCellAlignment) then    

    FOnGetCellAlignment(Self,ARow,ACol,HorAlignment,VerAlignment);

Importante: Antes de ativar um evento, é conveniente olhar primeiro se este evento tem um gerenciador de evento nomeado, como o usuário do componente não tem porque ter escrito este gerenciador. De lá a comparação  if Assigned: se há um gerenciador evento, ele é chamado, mas se não há nada é feito. 

<Figura: Bverde.gif - .325 K>Implementando a fonte, estilo e cor de celulas. 

Se você entendeu a seção anterior, não haverá nenhum problema para entender esta, como a forma de implementar a cor e a atribuição de fonte em uma certa celula será feita de um modo semelhante. 

Defininimos um novo evento que será ativado quando é necessário determinar os atributos da celula. Para isto, nós criaremos o tipo do evento, o campo privado armazenará isto e a propriedade associada a este evento: 

    

  TMultiGridColorEvent=procedure(Sender: TObject; ARow,Acol: LongInt;    

     AState: TGridDrawState; Abrush: TBrush; AFont:TFont) of object;    

  ...    

  FOnGetCellColor: TMultiGridColorEvent;    

  ...    

  property OnGetCellColor: TMultiGridColorEvent read FOnGetCellColor write FOnGetCellColor;

A diferença principal entre este evento e o correspondente ao alinhamento esta determinado pelos parâmetros ABruh e AFont. O usuário do componente devera retornar o brush e a Fonte a celula correspondente nas coordenadas ARow e ACol. O parâmetro AState nos informa do estado da cela (selected, focus, fixo...) 

<Figura: Bverde.gif - .325 K>Celulas Multi-linhas. 

Vamos passar agora para a implementação das celulas multi-linhas. 

Em primeiro lugar nós definiremos a interface. Para isto, nós criaremos uma propriedade nova chamada  MultiLinha . Esta propriedade será armazenada no campo FMultiLinha que será do tipo boolean. Se FMultiLinha é false, nosso componente se comportará como o StringGrid normal, enquanto se for true, se tornará um StringGrid Multi-Linhas. 

    

  private    

    FMultiLinha: Boolean;    

  ...    

    property MultiLinha: Boolean read FMultiLinha write SetMultiLinha default False;

Ele serve para fazer uma chamada ao método SetMultiLinha, se um valor novo é colocada para FMultiLinha o componente é redesenhado por meio da instrução o Invalidate. Esta mesma técnica é usada nos métodos SetAlignment, SetVerticalAlignment e SetColor. 

<Figura: Bverde.gif - .325 K>O coração do componente: o método DrawCell. 

Até agora, nós nos concentramos na interface de nosso componente; chegou o momento de definir o implementação. O processo inteiro de desenho de uma celula é feita através do método DrawCell. Este método é o verdadeiro coração de nosso componente, visto que é nele que deve ser escrito todo o código de calculo e desenho do texto correspondente em uma determinada celula. O evento  OnDrawCell passa os seguintes parâmetros ao método DrawCell: 

•ACol e ARow: Coordenadas da celula. •ARect: Estrutura de tipo retangular que identifica o canto superior esquerdo e o direito inferior (em pixels) da celula desejada. •AState: É o estado atual da celula (selecionda, enfocada ou fixa). Em princípio nós não usaremos o último valor neste parâmetro. 

Nós já sabemos onde. Agora falta ver como é. A princípio você pode ficar assustado com tudo aquilo nós temos que fazer: calcular o alinhamento horizontal e vertical, ativar os eventos de alinhamento e colorir, fragmenar o texto de uma celula em várias linhas... Mas não há nenhuma razão: o Delphi e as API do Windows nos ajudam a diminuir tudo isso para 20 ou 30 linhas compreensiveis. Logo o código que corresponde ao método DrawCell mostrado: 

    

procedure TMultiGrid.DrawCell(ACol,ARow : LongInt; ARect : TRect; AState : TGridDrawState);

Const

  TextAlignments : Array[TAlignment] of Word = (dt_Left, dt_Right, dt_Center);

Var

  HorAlignment : TAlignment;

  VerAlignment : TVerticalAlignment;

  Texto : string;

  Altura : integer;

  CRect : TRect;

  options : integer;

begin

  Texto:=Cells[ARow,Acol];

  HorAlignment:=FAlignment;

  VerAlignment:=FVerticalAlignment;

  if Assigned(FOnGetCellAlignment) then FOnGetCellAlignment(Self,ARow,ACol,HorAlineacion,VerAlignment);

  if Assigned(FOnGetCellColor) then

    FOnGetCellColor(Self,ARow,ACol,AState,Canvas.Brush,Canvas.Font);

  Canvas.FillRect(ARect);

  Inc(ARect.Left,2);

  Dec(ARect.Right,2);

  CRect:=Arect;

  options:=TextAlignments[HorAlignment] or dt_VCenter;

  if Multilinha then options:=options or dt_WordBreak;

  if not DefaultDrawing then

    inherited DrawCell(ACol,ARow,ARect,AState)

  else

    with ARect,Canvas do

    begin

      Altura:=DrawText(Handle,PChar(Texto),-1,CRect,options or dt_CalcRect);

      if FVerticalAlignment = vaCenter then

      begin

        if Altura < Bottom-Top+1 then

        begin

          Top:=(Bottom+Top-Altura) shr 1;

          Bottom:=Top+Altura;

        end;

      end

      else if FVerticalAlignment = vaBottom then

        if Altura < Bottom-Top+1 then Top:=Bottom-Altura;

      DrawText(Handle,PChar(Texto),-1,ARect,options)

    end;

end;

A primeira coisa que nós faremos é manter o conteúdo da celula para chamar a variável Texto . Logo os valores são determinados por padrão para as variáveis HorAlignment  e VerAlignment porque se o usuário não introduziu um alinhamento particular para a celula em questão estes alinhamentos padrões serão aplicados. 

Agora vem uma das chaves : a chamada para os eventos. Se o usuário escreveu um gerenciador para o evento de alinhamento, ele o chama. A mesma coisa acontece para o evento de Cor. Deste modo nós já temos o tratamento específico da celula. 

O proximo passo é desenhar o fundo da celula por meio do método FillRect para o qual nós passamos o requadro de desenho desta celula (ARect). 

Nós  faremos uma pausa agora para explicar agora como nós colocaremos o texto. 

Em princípio veja, a coisa lógica seria usar o método TextOut do objeto Canvas. Este método precisa como parâmetros as coordenadas (x ,y) onde colocar a string com o texto. Mas para nossos propósitos é ruim, pois teriamos que calcular a posição correta nas coordenadas (x,y). Nós também teríamos que calcular as divisões de palavras necessárias para as celulas multi-linhas, etc. Em resumo, é um rolo!! Mas nós podemos evitar todo este trabalho graças a uma função API do Windows: DrawText 

DrawText precisa dos seguintes parâmetros: 

•Um Handle para o objeto que nós queremos desenhar (o Canvas do grid). •Uma tipo de cadeia terminada em nulo com o texto a desenhar (Pchar(Texto)) •Um número que indica o número de caractere a desenhar (-1 para tudo) •O retângulo no qual o texto vai se enquadrar (TRect) •Uma série de opções de formatação de texto. Nós usaremos os controles de alinhamento de texto (dt_Left, dt_Center, dt_Right, dt_VCenter),  (dt_WordBreak) e o de cálculo de altura (dt_CalcRect) 

Há mais opções para o DrawText, se você quiser mais informações leia na ajuda on-line. 

Nós voltamos agora ao fluxo do programa. Depois de encher o fundo da celula com o FillRect, nós copiamos na variável CRect o retângulo original (ARect) e eles preparam as opções com que nós chamaremos o DrawText. Aqui surge um pequeno problema: o HorAlignment é do tipo TAlignment (ta_LeftJustify...) e DrawText não entende este tipo, por isso é necessária uma conversão entre este tipo e o do DrawText . Esta conversão é feita através de uma matriz constante denominado TextAlignments. Logo, se a propriedade Multilinha é true, dt_WordBreak é adicionado às opções do DrawText. 

O  é feito depois é verificar se o usuário trocou o valor da propriedade DefaultDrawing. Se o valor desta propriedade é false indica que o usuário se encarrega de todo o processo, caso contrario o componente se encarrega do desenho. 

Se o componente é encarregado de tudo, nós fazemos a primeira chamada ao DrawText para obter a altura exigida do retângulo da celula. Com esta altura,  sempre que possível (o multilinha ajusta o texto inteiro na celula), ele centraliza o texto na celula (ou no topo/rodapé). Uma vez feito isto, nós chamamos novamente o DrawText de forma que ele pega o lugar, e é colocado o texto no canvas do componente. Veja que para isso usamos ARect como retângulo nesta ocasião. 

Estas são as grandes caracteristicas com a operação do método DrawCell. 

<Figura: Bverde.gif - .325 K>Outros detalhes. 

Por ultimo, quero mencionar alguns detalhes pequenos: 

•As propriedades de Opções do StringGrid padrão são redefinidas por padrão para true e as opções goRowSizing e goColSizing para multilinhas do nosso componente true por padrão. •Como sempre, no construtor nós declaramos os valores padrões defeito para as propriedades diferentes que nós definimos (alinhamento, cor, multilinha). •Depois no codigo fonte é mostrado um exemplo dos eventos de alinhamento, cores e multilinha. 

<Figura: Bverde.gif - .325 K>Codigo Fonte do Componente. 

    

unit MultiGrid;       { (c) 1997 by Luis Roche }

interface

uses

  SysUtils, WinTypes, WinProcs, Messages, Classes, Graphics, Controls,

  Forms, Dialogs, Grids;

type

  TVerticalAlignment = (vaTop, vaCenter, vaBottom);

  TMultiGridAlignmentEvent=procedure(Sender:TObject;ARow,ACol:LongInt;var HorAlignment:TAlignment;var VerAlignment:TVerticalAlignment) of object;

  TMultiGridColorEvent=procedure(Sender:TObject;ARow,Acol:LongInt;AState:TGridDrawState;ABrush:TBrush;AFont:TFont) of object;

  TMultiGrid = class(TStringGrid)

  private

    FAlignment : TAlignment;

    FVerticalAlignment : TVerticalAlignment;

    FMultiLinha : Boolean;

    FOnGetCellAlignment : TMultiGridAlignmentEvent;

    FOnGetCellColor : TMultiGridColorEvent;

    procedure SetAlignment(Valor : TAlignment);

    procedure SetVerticalAlignment(Valor : TVerticalAlignment);

    procedure SetMultiLinha(Valor : Boolean);

  protected

    procedure DrawCell(ACol,ARow : LongInt; ARect : TRect; AState : TGridDrawState); override;

  public

    constructor Create(AOwner : TComponent); override;

  published

    property Alignment : TAlignment read FAlignment write SetAlignment default taLeftJustify;

    property VerticalAlignment : TVerticalAlignment read FVerticalAlignment write SetVerticalAlignment default vaCenter;

    property MultiLinha : Boolean read FMultiLinha write SetMultiLinha default False;

    property OnGetCellAlignment : TMultiGridAlignmentEvent read FOnGetCellAlignment write FOnGetCellAlignment;

    property OnGetCellColor : TMultiGridColorEvent read FOnGetCellColor write FOnGetCellColor;

    property Options default [goFixedVertLine,goFixedHorzLine,goVertLine,goHorzLine,goRangeSelect,goRowSizing,goColSizing];

  end;

procedure Register;

implementation

constructor TMultiGrid.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  FAlignment:=taLeftJustify;

  FVerticalAlignment:=vaCenter;

  {FColor:=clWindowText;}

  FMultiLinha:=False;

  Options:=[goFixedVertLine,goFixedHorzLine,goVertLine,goHorzLine,goRangeSelect,goRowSizing,goColSizing];

end;

procedure TMultiGrid.SetAlignment(Valor : TAlignment);

begin

  if valor <> FAlignment then

  begin

    FAlignment:=Valor;

    Invalidate;

  end;

end;

procedure TMultiGrid.SetVerticalAlignment(Valor : TVerticalAlignment);

begin

  if valor <> FVerticalAlignment then

  begin

    FVerticalAlignment:=Valor;

    Invalidate;

  end;

end;

procedure TMultiGrid.SetMultiLinha(Valor : Boolean);

begin

  if valor <> FMultiLinha then

  begin

    FMultiLinha:=Valor;

    Invalidate;

  end;

end;

procedure TMultiGrid.DrawCell(ACol,ARow : LongInt; ARect : TRect; AState : TGridDrawState);

Const

  TextAlignments : Array[TAlignment] of Word = (dt_Left, dt_Right, dt_Center);

Var

  HorAlignment : TAlignment;

  VerAlignment : TVerticalAlignment;

  Texto : string;

  Altura : integer;

  CRect : TRect;

  Options : integer;

begin

  Texto:=Cells[ARow,Acol];

  HorAlignment:=FAlignment;

  VerAlignment:=FVerticalAlignment;

  if Assigned(FOnGetCellAlignment) then FOnGetCellAlignment(Self,ARow,ACol,HorAlignment,VerAlignment);

  if Assigned(FOnGetCellColor) then

    FOnGetCellColor(Self,ARow,ACol,AState,Canvas.Brush,Canvas.Font);

  Canvas.FillRect(ARect);

  Inc(ARect.Left,2);

  Dec(ARect.Right,2);

  CRect:=Arect;

  Options:=TextAlignments[HorAlignment] or dt_VCenter;

  if Multilinha then Options:=Options or dt_WordBreak;

  if not DefaultDrawing then

    inherited DrawCell(ACol,ARow,ARect,AState)

  else

    with ARect,Canvas do

    begin

      Altura:=DrawText(Handle,PChar(Texto),-1,CRect, options or dt_CalcRect);

      if FVerticalAlignment = vaCenter then

      begin

        if Altura < Bottom-Top+1 then

        begin

          Top:=(Bottom+Top-Altura) shr 1;

          Bottom:=Top+Altura;

        end;

      end

      else if FVerticalAlignment = vaBottom then

        if Altura < Bottom-Top+1 then Top:=Bottom-Altura;

      DrawText(Handle,PChar(Texto),-1,ARect,options)

    end;

end;

procedure Register;

begin

  RegisterComponents('Curso', [TMultiGrid]);

end;

end.

<Figura: Bverde.gif - .325 K>  Exemplo de utilização. 

  

Logo um exemplo de utilização do nosso componente novo:

procedure TForm1.FormCreate(Sender: TObject);

begin

  MultiGrid1.Cells[0,1]:='Janeiro';

  MultiGrid1.Cells[0,2]:='Fevereiro';

  MultiGrid1.Cells[0,3]:='Total Ano';

  MultiGrid1.Cells[0,4]:='Notas';

  MultiGrid1.Cells[1,0]:='Área A';

  MultiGrid1.Cells[2,0]:='Área B';

  MultiGrid1.Cells[3,0]:='Outras Areas';

  MultiGrid1.Cells[4,0]:='TOTAL';

  MultiGrid1.Cells[1,1]:='1.000.000';

  MultiGrid1.Cells[1,2]:='1.250.000';

  MultiGrid1.Cells[1,3]:='9.150.000';

  MultiGrid1.Cells[1,4]:='Incremento sobre ano anterior';

  MultiGrid1.Cells[2,1]:='1.450.000';

  MultiGrid1.Cells[2,2]:='  950.000';

  MultiGrid1.Cells[2,3]:='4.150.000';

  MultiGrid1.Cells[2,4]:='Decremento';

  MultiGrid1.Cells[3,1]:='4.000.000';

  MultiGrid1.Cells[3,2]:='3.250.000';

  MultiGrid1.Cells[3,3]:='17.250.000';

  MultiGrid1.Cells[3,4]:='Incremento sobre ano anterior';

  MultiGrid1.Cells[4,1]:='6.450.000';

  MultiGrid1.Cells[4,2]:='5.450.000';

  MultiGrid1.Cells[4,3]:='30.550.000';

  MultiGrid1.Cells[4,4]:='';

end;

procedure TForm1.Button1Click(Sender: TObject);

begin

  MultiGrid1.Color:=clRed;

end;

procedure TForm1.MultiGrid1GetCellAlignment(Sender: TObject; ARow,

  ACol: Longint; var HorAlignment: TAlignment;

  var VerAlignment: TVerticalAlignment);

begin

  if (ACol in [1..3]) and (ARow in [1..4]) then

    HorAlignment:=taRightJustify

  else HorAlignment:=taCenter;

end;

procedure TForm1.MultiGrid1GetCellColor(Sender: TObject; ARow,

  Acol: Longint; AState: TGridDrawState; Abrush: TBrush; AFont: TFont);

begin

  if (ARow=0) then

  begin

    ABrush.Color:=clMaroon;

    AFont.Color:=clWhite;

    AFont.Style:=[fsBold];

  end

  else if (ACol=0) then

  begin

    AFont.Color:=clBlack;

    AFont.Style:=[fsBold];

    ABrush.Color:=clYellow;

  end

  else

  begin

    AFont.Color:=clBlack;

    ABrush.Color:=clYellow;

  end;

end;

Curso de criação de componentes em Delphi

  Unidade 7. TDBViewer: Um visualizador rápido de bancos de dados.  <Figura: delphi.gif - 582,0 K> 

------------------------------------------------------------------------

<Figura: dbviewer.gif>Nesta unidade nós transformaremos uma forma em um componente. Em resumo, nós criaremos uma forma para visualizar bancos de dados rapidamente e nós integraremos isto dentro de um componente. Deste modo em nossos projetos, quando queremos visualizar um banco de dados rapidamente, chamamos um método de nosso componente para realiza-lo. 

Esta conversão de caixas de diálogo em componentes é uma opção muito poderosa que nos permite reutilizar o código de nossos projetos diversos de um modo rápido e simples. Sem dúvida assim que nós saibamos a técnica para alcançar-lo nós teremos possibilidades infinitas para converter form's das que nós usamos freqüentemente em componentes. Seria bastante dizer que eu tenho um componente que encapsula uma aplicação inteira de 10 ou 12 janelas!. 

É necessario dizer que algumas funções que usaremos aqui não existem no Delphi 1; se você for usar o Delphi 1 entao é melhor procurar na ajuda on-line algo semelhante ao que nos vamos usar. 

<Figura: Bverde.gif - .325 K>Objetivo do componente 

Nosso objetivo é criar um componente que mostrara uma janela ao usuário através da definição de propriedades e chamadas de métodos em design-time para que seja ativado em run-time. 

A forma que nós vamos integrar no componente é visualizador / editor de  bancos de dados. Logo o componente é mostrado em run-time e suas funções principais é comentado. 

<Figura>

•Por meio dos dois combo boxes o usuário define o alias e o nome da tabela a ser visualizado. •Um dbgrid é mostrado para a consulta e edição dos dados. •Em um painel adicional ele podera fazer consultas e buscas por palavras em qualquer campo de um banco de dados. •

O grid tem um menu  popup que permite ou não o filtro de registros. 

Criação do Componente

•Primeiro, nós criaremos o forma de nosso componente. •Depois criaremos o componente, inclusive propriedades, métodos, etc. •Por ultimo nós faremos a conexão entre o componente e a forma. 

<Figura: Bverde.gif - .325 K>  O projeto da forma 

A criação da forma sera feita do modo tradicional, quer dizer, no Delphi, nós faremos isso em  File|New e escolhemos a forma em branco. 

A seção de interface da forma é a seguinte: 

    

unit frmView;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  StdCtrls, DB, DBTables, Grids, DBGrids, ExtCtrls, Buttons, Menus;

type

  TfrmVisor = class(TForm)

    tTable: TTable;

    dsDataSource: TDataSource;

    dbGrid: TDBGrid;

    pTop: TPanel;

    pSelectTable: TPanel;

    pBusca: TPanel;

    gbTables: TGroupBox;

    Alias: TLabel;

    Table: TLabel;

    cbAlias: TComboBox;

    cbTables: TComboBox;

    gbBusca: TGroupBox;

    cbFields: TComboBox;

    eSearch: TEdit;

    Buscar: TSpeedButton;

    BuscarSeguinte: TSpeedButton;

    PopupMenu1: TPopupMenu;

    Igual1: TMenuItem;

    Distinto1: TMenuItem;

    Menor1: TMenuItem;

    Maior1: TMenuItem;

    N1: TMenuItem;

    Ativar1: TMenuItem;

    Eliminar1: TMenuItem;

    procedure FormCreate(Sender: TObject);

    procedure FillcbFields;

    procedure Inicializar;

    procedure cbAliasChange(Sender: TObject);

    procedure cbTablesChange(Sender: TObject);

    procedure BuscarClick(Sender: TObject);

    procedure BuscarSeguinteClick(Sender: TObject);

    procedure dbGridColEnter(Sender: TObject);

    procedure PopupMenu1Popup(Sender: TObject);

    procedure Ativar1Click(Sender: TObject);

    procedure Eliminar1Click(Sender: TObject);

    procedure Igual1Click(Sender: TObject);

    procedure Distinto1Click(Sender: TObject);

    procedure Menor1Click(Sender: TObject);

    procedure Maior1Click(Sender: TObject);

    private

  public

    FReadOnly, FFiltrar, FBuscar : boolean;

  end;

procedure PascalStr(var s : string);

procedure ComaStr(var s : string);

function Condicao(AField : TField; AOp : string) : string;

procedure AddFilter(AField : TField; AOp : string);

var

  frmVisor: TfrmVisor;

Seção Pubic: Nela nós definimos três variáveis de tipo boolean: FReadOnly, FFiltrar e FBuscar. Estas três variáveis formam a interface entre a forma e o componente. Deste modo, à propriedade ReadOnly que nós implementaremos no componente a variável FReadOnly  da forma  correspondente e, de um modo semelhante acontece com as propriedades (AllowSearch) e (AllowFilter) do componente. Será o componente, e não a forma que executará os métodos. 

O método  de Inicializar é o que se encarrega de ajustar a visualização da forma de acordo com a variável FReadOnly , FFiltrar e FBuscar. Este método será chamado uma vez pelo próprio componente uma vez que o valor das variáveis forem incluidas: 

•Ajusta a propriedade ReadOnly do grid de acordo com o valor de FReadOnly. •Mostra ou não o painel de procura de acordo com o valor de FBuscar. •Coloca ou não o menu popup para o grid de acordo com o valor de FFiltrar. 

Veremos agora como é feita a visualização . No evento OnCreate da forma você começa a encher o combobox cbAlias com todos os aliases possiveis. Depois você tera que encher o Combobox  cbTables com todas as tabelas existentes no alias com o método (cbAliasChange). Este metodo força a chamada para o método cbTablesChange , o qual se encarrega de ativar o quadro (ele também enche o ComboBox cbFields com os campos da tabela selecionada). Deste modo, o grid que já é conectado ao datasource e a tabela correspondente. Mais tarde se o usuário selecionar outro alias ou tabela, o processo se  repete e a nova tabela selecionada é mostrada. 

Um tratamento de erros na abertura das tabelas é recomendável, porem este não é o objetivo desta lição. Mas você poderia colocar no mínimo um bloco Try... Except. 

Nós veremos agora como fazer as buscas de registros. Para isto nós usaremos o método BuscarClick que procurará a string do Edit eSearch no campo definido pelo combobox cbFields.. A procura é feita através dos métodos do objeto TTable: FindFirst e FindNext que eu suponho que você saiba mais do que o suficiente. Em todo caso, você achará informação sobre o mesmo na ajuda on-line de Delphi. 

Agora implementaremos o filtro de registros. Há diversas formas de fazer isto, e optei por usar um menu popoup que permite ativar, desativar e adicionar condições novas para o filtro. Sendo mais claro o menu tem as seguintes opções: Adicionar uma condição do tipo Campo = valor, Campo > Valor, Campo < valor e Campo <> Valor, ativar o filtro e eliminar o filtro. As condições diversas do filtro se unem  um ao outro por meio do operador AND. 

O método Activar1Click e Eliminar1Click não precisam de muita explicação desde que eles são limitados para pôr a propriedade Filtered do objeto TTable True ou False como corresponde. 

Os métodos restantes do menu (Igual1Click...) eles adicionam uma nova condição para o filtro. Para isto, o procedimento AddFilter é usado e  recebe como parâmetros o campo onde vai ser colocado o filtro e sua condição do valor (isso é a coluna ativa do dbgrid quando é apertado o botão direito do mouse) e o operador (=, <,> ou <> ). Este método, depois de formatado corretamente, a string é colocada no filtro do objeto TTable, que faz um refresh na Tabela e é ativado o Filtro. 

E com isto nós terminamos a parte que corresponde ao desenvolvimento da forma. Fácil, não é? De qualquer maneira, se você tem alguma dúvida, o codigo fonte esta logo abaixo. <Figura: Bverde.gif - .325 K> Criação do componente 

Vamos trabalhar agora com a criação do componente, o que é muito simples. Consiste em três propriedades e dois métodos, que a esta altura do curso já não devem ser problema para você. 

•A propriedade  ReadOnly determinará se o grid da forma serão editaveis ou não. Para ele o valor padrão é false, quer dizer, editavel. Os métodos read e write se limitam a ler e escrever no campo FReadOnly do componente (não confundir com o FReadOnly da forma visto na seção anterior). •A propriedade  AllowSearch determina se o painel de procura deve aparecer na forma. Por padrão o valor dele deve ser True. Os métodos deles são lidos e escritos diretamente no campo FAllowSearch. •A propriedade AllowFilter determina se é permitido ou não o filtro de registros na forma. Por padrão o valor dele é true. Os métodos são iguais ao anterior! ;) •O contrutor é limitado a chamar o contrutor herdado e nomear os valores padrões  para as propriedades. •O método Execute executara o componente nos mostrando a forma. 

E isso é tudo. Simples ou não? Esta simplicidade será em geral sempre que você  transformar uma forma em componente. Basta criar as propriedades que agirão na forma e definir um método que chama a forma e acabou. 

Ah! que você não pode esquecer de colocar o nome da sua unidade da forma na clausula uses do componente (na seção de implementação) a unidade da forma. Em nosso caso: 

...

implementation

uses frmView;

...

<Figura: Bverde.gif - .325 K>A conexão entre o componente e a forma 

Como nós há pouco dissemos, a execução da forma é levada através do método Execute do componente: 

procedure TDBViewer.Execute;

begin

  {Criaçao do Form}

  frmVisor:=TFrmVisor.Create(Application);

  try

    {Colocar os valores das propriedades do componente nas variaveis 

     correspondentes do form}

    frmVisor.FReadOnly:=FReadOnly;

    frmVisor.FBuscar:=FAllowSearch;

    frmVisor.FFiltrar:=FAllowFilter;

    {Inicializar o form}

    frmVisor.Initialize;

    {Mostrar a form em modal}

    frmVisor.ShowModal;

  finally

    {Liberar o form}

    frmVisor.Free;

  end;

end;

O método começa a criar a forma cujo o proprietário será a aplicação. Depois ele é levado a conexão entre a forma e o componente. Porque eles partem dos valores colocados para as variáveis  FReadOnly , FBuscar e FFiltrar da forma  correspondendo ao componente. Uma vez feita esta tarefa, nós deveremos fazer com que a forma seja visualizada corretamente baseado nestes valores (lembre-se que a forma foi criada, mas não foi mostrado ainda). Isto que chama ao método para Inicializar da forma é adquirido (você viu este método no desenvolvimento da forma). 

Nós só deixamos mostrar a forma de um modo modal e esperar que o usuário se feche esta forma; momento no o qual nós destruiremos a forma por meio da chamada para o método Free. 

Como último aspecto a destacar temos que avisá-lo que o bloco try... finally nos assegura que os processo foi feito corretamente e que a memória ocupada sera liberada. 

<Figura: Bverde.gif - .325 K>Instruções de uso do componente e conclusões 

Uma vez registrado o componente na paleta do modo habitual, o uso dele é muito simples. Quando nós queremos usar isto em um projeto, basta colocar o componente DBViewer onde nós quisermos, colocar os valores nas propriedades e chamar o método Execute. 

Como conclusão, nós pudemos ver que é mais simples do que parece transformar nossas formas em componentes e chama-los pelo método execute. Este componente que nós fizemos foi bem simplificado; espero que você coloque novas funções nela ao longo do tempo. Aqui vao algumas idéias que podem ser colocados depois OnOpenTable, OnTableError, opção de impressao dos dados, etc. Use sua imaginação conforme sua necessidade. 

<Figura: Bverde.gif - .325 K>Codigo Fonte  do componente. 

    

unit DBView;          { (c) 1997 by Luis Roche }

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs;

type

  TDBViewer = class(TComponent)

  private

    FReadOnly : boolean;

    FAllowSearch : boolean;

    FAllowFilter : boolean;

  protected

  public

    constructor Create(AOwner : TComponent); override;

    procedure Execute;

  published

    property ReadOnly : boolean read FReadOnly write FReadOnly default False;

    property AllowSearch : boolean read FAllowSearch write FAllowSearch default True;

    property AllowFilter : boolean read FAllowFilter write FAllowFilter default True;

  end;

procedure Register;

implementation

uses frmView;

constructor TDBViewer.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  FReadOnly:=False;

  FAllowSearch:=True;

  FAllowFilter:=True;

end;

procedure TDBViewer.Execute;

begin

  frmVisor:=TFrmVisor.Create(Application);

  try

    frmVisor.FReadOnly:=FReadOnly;

    frmVisor.FBuscar:=FAllowSearch;

    frmVisor.FFiltrar:=FAllowFilter;

    frmVisor.Inicializar;

    frmVisor.ShowModal;

  finally

    frmVisor.Free;

  end;

end;

procedure Register;

begin

  RegisterComponents('Curso', [TDBViewer]);

end;

end.

<Figura: Bverde.gif - .325 K>Codigo fonte da forma. 

unit frmView;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  StdCtrls, DB, DBTables, Grids, DBGrids, ExtCtrls, Buttons, Menus;

type

 TfrmVisor = class(TForm)

    tTable: TTable;

    dsDataSource: TDataSource;

    dbGrid: TDBGrid;

    pTop: TPanel;

    pSelectTable: TPanel;

    pBusca: TPanel;

    gbTables: TGroupBox;

    Alias: TLabel;

    Table: TLabel;

    cbAlias: TComboBox;

    cbTables: TComboBox;

    gbBusca: TGroupBox;

    cbFields: TComboBox;

    eSearch: TEdit;

    Buscar: TSpeedButton;

    BuscarSeguinte: TSpeedButton;

    PopupMenu1: TPopupMenu;

    Igual1: TMenuItem;

    Distinto1: TMenuItem;

    Menor1: TMenuItem;

    Maior1: TMenuItem;

    N1: TMenuItem;

    Ativar1: TMenuItem;

    Eliminar1: TMenuItem;

    procedure FormCreate(Sender: TObject);

    procedure FillcbFields;

    procedure Inicializar;

    procedure cbAliasChange(Sender: TObject);

    procedure cbTablesChange(Sender: TObject);

    procedure BuscarClick(Sender: TObject);

    procedure BuscarSeguinteClick(Sender: TObject);

    procedure dbGridColEnter(Sender: TObject);

    procedure PopupMenu1Popup(Sender: TObject);

    procedure Ativar1Click(Sender: TObject);

    procedure Eliminar1Click(Sender: TObject);

    procedure Igual1Click(Sender: TObject);

    procedure Distinto1Click(Sender: TObject);

    procedure Menor1Click(Sender: TObject);

    procedure Maior1Click(Sender: TObject);

    private

  public

    FReadOnly, FFiltrar, FBuscar : boolean;

  end;

procedure PascalStr(var s : string);

procedure ComaStr(var s : string);

function Condicao(AField : TField; AOp : string) : string;

procedure AddFilter(AField : TField; AOp : string);

var

  frmVisor: TfrmVisor;  

implementation

{$R *.DFM}

procedure TfrmVisor.FormCreate(Sender: TObject);

begin

  Session.GetAliasNames(cbAlias.Items);

  cbAlias.ItemIndex:=0;

  cbAliasChange(Sender);

  BuscarSeguinte.Enabled:=False;

end;

procedure TfrmVisor.Inicializar;

begin

  dbGrid.ReadOnly:=FReadOnly;

  pBuscar.Visible:=FBuscar;

  if FFiltrar then

    dbGrid.PopupMenu:=PopupMenu1

  else

    dbGrid.PopupMenu:=nil;

end;

procedure TfrmVisor.cbAliasChange(Sender: TObject);

begin

  Session.GetTableNames(cbAlias.Items[cbAlias.ItemIndex],'*.*',True,True,cbTables.Items);

  cbTables.ItemIndex:=0;

  cbTablesChange(Sender);

end;

procedure TfrmVisor.cbTablesChange(Sender: TObject);

begin

  tTable.Active:=False;

  tTable.DatabaseName:=cbAlias.Text;

  tTable.TableName:=cbTables.Text;

  tTable.Active:=True;

  FillcbFields;

end;

procedure TfrmVisor.FillcbFields;

Var

  i : integer;

begin

  cbFields.Items.Clear;

  cbFields.Text:='';

  for i:=0 to tTable.FieldCount-1 do

    cbFields.Items.Add(tTable.Fields[i].DisplayLabel);

end;

procedure TfrmVisor.BuscarClick(Sender: TObject);

Var

  F : TField;

  s : string;

begin

  if cbFields.ItemIndex=-1 then

    exit;

  with tTable do

  begin

    s:=eSearch.Text;

    F:=Fields[cbFields.ItemIndex];

    case F.DataType of

      FtString, FtDate, FtTime, FtDateTime : PascalStr(s);

      FtFloat : ComaStr(s);

    end;

    Filter:='['+F.FieldName+']='+s;

    FindFirst;

    BuscarSeguinte.Enabled:=Found;

  end;

end;

procedure TfrmVisor.BuscarSeguinteClick(Sender: TObject);

begin

  tTable.FindNext;

  BuscarSeguinte.Enabled:=tTable.Found;

end;

procedure PascalStr(var s : string);

Var

  i : integer;

begin

  for i:=Length(s) downto 1 do

    if s[i]='''' then Insert('''',s,i);

  s:=''''+s+'''';

end;

procedure ComaStr(var s : string);

Var

  i : integer;

begin

  for i:=Length(s) downto 1 do

    if s[i]=',' then s[i]:='.';

end;

procedure TfrmVisor.dbGridColEnter(Sender: TObject);

begin

  cbFields.ItemIndex:=dbGrid.SelectedField.Index;

end;

procedure TfrmVisor.PopupMenu1Popup(Sender: TObject);

begin

  with tTable do

  begin

    Ativar1.Checked:=Filtered;

    Ativar1.Enabled:=Filter<>'';

    Eliminar1.Enabled:=Filter<>'';

  end;

end;

procedure TfrmVisor.Ativar1Click(Sender: TObject);

begin

  tTable.Filtered:=not tTable.Filtered;

  tTable.Refresh;

end;

procedure TfrmVisor.Eliminar1Click(Sender: TObject);

begin

  tTable.Filtered:=False;

  tTable.Filter:='';

  tTable.Refresh;

end;

procedure TfrmVisor.Igual1Click(Sender: TObject);

begin

  AddFilter(dbGrid.SelectedField,'=');

end;

procedure TfrmVisor.Distinto1Click(Sender: TObject);

begin

  AddFilter(dbGrid.SelectedField,'<>');

end;

procedure TfrmVisor.Menor1Click(Sender: TObject);

begin

  AddFilter(dbGrid.SelectedField,'<=');

end;

procedure TfrmVisor.Maior1Click(Sender: TObject);

begin

  AddFilter(dbGrid.SelectedField,'>=');

end;

procedure AddFilter(AField : TField; AOp : string);

Var

  s : string;

begin

  s:=Condicao(AField,AOp);

  with AField.DataSet do

  begin

    if Filter='' then Filter:=s

    else Filter:=Filter+' AND ' + s;

    Filtered:=True;

    Refresh;

  end;

end;

function Condicao(AField : TField; AOp : string) : string;

Var

  Valor : string;

begin

  Valor:=AField.AsString;

  case AField.DataType of

    FtString, FtDate, FtTime, FtDateTime : PascalStr(Valor);

    FtFloat : ComaStr(Valor);

  end;

  Condicao:=Format('([%s] %s %s)',[AField.FieldName,AOp,Valor]);

end;

end.

Curso de criação de componentes em Delphi   

Unidade 8. Editores de Propriedades (I).  <Figura: delphi.gif - 582,0 K> 

------------------------------------------------------------------------

Nesta unidade nós aprenderemos a criar nossos próprios editores de propriedades. Os editores de propriedades são uma ferramenta poderosa que fazem diferença entre um componente simplesmente aceitável e um realmente bom. Por eles nós poderemos dotar a nossos componentes de características novas que vão além da validação do valor das propriedades. Graças a eles nós faremos a vida mais fácil para os programadores que usam nossos componentes, e isso sempre é bom não é? 

<Figura: Bverde.gif - .325 K>Introdução aos editores de propriedades 

Até agora nós se criamos os componentes sem se preocupar como foram introduzidos os valores das diferentes propriedades definidas neles. Chegou o momento de estudarmos este aspecto. Como já sabemos, para inserir e ler o valor das propriedades em design-time nós usamos o Object Inspector. O Object Inspector se divide em duas colunas: no da direita estão os nomes das propriedades e no da esquerda seus valores. Introduzimos um valor novo e ele aparece. Mas isto é só aparência. Os componentes se interagem com o Object inspector através dos editores de propriedades. Do ponto de vista do usuário, o Object Inspector é o responsável de publicar as propriedades, mas por trás  há uma série de objetos, os editores de propriedades que se encarregam de definir as capacidades de edição das diversas propriedades de um componente. 

Em design-time, quando você seleciona um componente, o Object Inspector cria instâncias dos editores de propriedades necessários para publicar as propriedades definidas no componente selecionado. Quando termina a edição, o mesmo Object Inspector destroi os editores de propriedades criados. 

O Delphi tem uma série de editores de propriedades padrões que são o bastante para a maioria das propriedades com que nós habitualmente trabalhamos. Mas como sempre, o Delphi é tão potente que permite que nós criemos nossos próprios editores que podem substituir certas propriedades que estão por padrão. Pobre VB! ;) 

<Figura: Bverde.gif - .325 K>Os deveres do Editor de Propriedades 

O Editor de propriedades deve dar resposta a dois aspectos principais: 

•Definir como a propriedade devera ser publicada. Aqui há duas possibilidades: pode ser modificado o valor da propriedade no próprio Object Inspector  ou você pode usar uma caixa de diálogo para adicionar mais flexibilidade à edição (p.e. propriedade cor) •Converter o valor da propriedade para um valor tipo string. O Object Inspector sempre trabalha com Strings. Embora a propriedade seja do tipo inteiro ou Ponto Flutuante, o  Object Inspetor só usa o tipo String em suas propriedades. É o editor de propriedades que faz todo trabalho na representação desta propriedade. Traduzindo:  pode ir da coisa mais simples (usar a função IntToStr) até a uma mais complexa (programar uma rotina para a conversões). Tudo dependerá do tipo de propriedade com que nós estamos trabalhando. 

<Figura: Bverde.gif - .325 K>Passos necessários para escrever um editor de propriedades 

Os passos necessários para escrever um editor de propriedades são o seguinte: 

•Criar uma nova unidade na qual nós definiremos o editor de propriedades. Mais tarde nós falaremos mais amplamente sobre este ponto, desde que não é tão trivial quanto pode parecer em princípio;) •Colocar a unidade DsgnIntf para a clausula uses do editor de propriedades. Nesta unidade estão definidos os editores de propriedades padrões que o Delphi usa, além da importante classe TPropertyEditor,que é a base de todos os editores de propriedades. •Criar uma classe nova que descende de TPropertyEditor ou de algum do descendentes deles. Por convenção, o nome dos editores de propriedades terminam com a palavra Property. P.e. TIntegerProperty, TStringProperty... Para os editores principais de propriedades logo são mostrados por padrão no Delphi. •Editor de propriedades Tipo TPropertyEditor Bases de classe para todos os editores de propriedades TIntegerProperty Byte, word, integer, Longint, TCharProperty Char TEnumProperty Tipos enumerados TSetProperty Sets TFloatPropertySingle, Doble, Extended, Comp, Currency, TStringProperty StringsTClassProperty Qualquer objeto TMethodProperty Qualquer método (eventos) TComponentProperty Para propriedades que fazem referência a componentes 

  

•Implementar os métodos necessários para adicionar ao editor de propriedades as funcionalidades queridas. •Registrar o editor de propriedades na VCL 

Mais tarde nós nos aprofundaremos em todos os aspectos dos outros editores de propriedades, vamos criar agora o nosso primeiro editor de propriedades. 

<Figura: Bverde.gif - .325 K>Editor de propriedades para números binários 

<Figura>

Nós desenvolveremos nosso primeiro editor de propriedades que será editavel no próprio Object Inspector. Imagine que nós criamos um componente com uma propriedade do tipo inteiro que tenha um valor para ser visualizado em base binária e não em base decimal.. Se nós não criássemos editor de propriedades e deixássemos que o editor de propriedades padrão do Delphi (TIntegerProperty no caso) fizesse o trabalho a visualização seria em base decimal e não em base binária como nós queremos.. Isto é típico caso no qual desenvolvendo um editor de propriedades simples se economiza muito trabalho para o futuro usuário do nosso componente e evita que ele tenha que fazer a conversão entre decimal e binário. 

Como nós já comentamos, nós temos que decidir que de classe se derivará o nosso editor. Em nosso caso é fácil, a propriedade armazenará um valor do tipo inteiro, assim nós usaremos IntegerProperty 

Muito bem, nossa propriedade armazena um inteiro, mas nós não queremos que o Object Inspector nos mostre diretamente o valor decimal dele, nós queremos fazer uma conversão deste valor de decimal a base binária e que só então o Object Inspetor nos mostrará o valor. Para fazer isto nós implementaremos  (override) na função GetValue. Esta função, definida em TPropertyEditor,  é chamada pelo Object Inspector para obter a representação do valor da propriedade em forma de string. Dentro desta função nós deveremos converter o valor da propriedade decimal para binário e depois  transformar este valor em String, pois o Object Inspector só mostra Strings. Deste modo a função GetValue fica assim: 

    

unit BinaryPropEd;

interface

uses DsgnIntf;

type

  TBinIntegerProperty = class(TIntegerProperty)

  public

    function GetValue : string; override;

    procedure SetValue(const Value : String); override;

  end;

procedure Register;

implementation

Const

  Bits16 : Array [1..16] of Integer = (32768,16384,8192,4096,2048,1024,512,256,128,64,32,16,8,4,2,1);

function TBinIntegerProperty.GetValue : string;

Var

  Num, i : integer;

begin

  Num:=GetOrdValue;

  Result := '0000000000000000';

  for i := 1 to 16 Do

    if ( Num >= Bits16[i] ) Then

    begin

      Num := Num - Bits16[i];

      Result[i] := '1';

    end;

  if ( Num > 0 ) Then

    raise EPropertyError.Create('Error converting '+IntToStr(GetOrdValue) + ' to binary');

  Insert('B',Result,1);  

end;

...

end.

Da implementação desta função a parte mais importante é a primeira linha: 

Num:=GetOrdValue

Nós precisamos obter o valor que está naquele momento na propriedade para trabalhar nele. Porque nós usamos isto o método GetOrdValue, definido novamente em TPropertyEditor, o qual  se encarrega de devolver o valor da propriedade em forma de ordinal (inteiro). De um modo semelhante, existem os métodos GetFloatValue, GetMethodValue, GetVarValue, etc. Para usar com o tipo de propriedade correspondente. 

Uma vez armazenado o valor da propriedade na variável  Num , a conversão começa do valor decimal para binário, o qual é fácil de entender. É bom saber que o número máximo de dígitos binários é 16, margem mais do que suficiente para a maioria das aplicações. 

Por Ultimo nós temos que devolver um valor do tipo String como resultado da função. Porque nós vamos armazenar isto em uma propriedade que aceita somente String. 

Para concluir, nós colocamos antes da cadeia de string uma letra ' B' para indicar que é base binária. 

E isso é tudo  que esta função faz. Deste modo, quando o Object Inspetor , ele chamará o método GetValue o  qual lhe devolverá o string correspondente. Mas nós partimos para outro problema: seria bom se pudessemos introduzir o valor da propriedade tanto em decimal como em binário, não é mesmo? 

Para adquirir esta funcionalidade nós devemos implementar (override) do método SetValue, definido na classe TPropertyEditor. Quando o usuário entra um valor novo no  Object Inspetor, ele chama o método SetValue, o qual devera fazer a tradução inversa ao feito pelo método GetValue. Quer dizer, deveria converter o string que contém o valor novo da propriedade para o tipo de dados desta propriedade. Em nosso caso, o string entrará em base decimal ou binário (neste último caso, estará a primeira letra da cadeia ' B') e converter isto para um decimal. Porque nós implementaremos isto o método SetValue do modo seguinte: 

...

type

  TBinIntegerProperty = class(TIntegerProperty)

  public

    function GetValue : string; override;

    procedure SetValue(const Value : String); override;

  end;

procedure Register;

implementation

...

procedure TBinIntegerProperty.SetValue(const Value : String);

Var

  i, Total, Longitud : integer;

  NumText : string;

begin

  if UpperCase(Value[1])='B' then

  begin

    NumText:=Copy(Trim(Value),2,Length(Trim(Value))-1);

    NumText:=Copy('0000000000000000',1,16-Length(NumText)) + NumText;

    Total:=0;

    for i:=1 to Length(NumText) do

    begin

      if not (NumText[i] in ['0','1']) then

        raise EPropertyError.Create(NumText[i] + ' is not a valid binary digit')

      else if NumText[i]='1' then

        Total:=Total+Bits16[i];

    end;

    SetOrdValue(Total);

  end

  else

    SetOrdValue(StrToInt(Value));

end;

...

end.

Na implementação deste primeiro método conferimos nós se o usuário introduziu o valor novo da propriedade em base decimal ou em base binária. No primeiro caso, é só necessário converter o string para inteiro por meio da função StrToInt(Value) e, depois, usar o método SetOrdValue para armazenar o valor correspondente. De um modo semelhante, de acordo com o tipo da propriedade, existem os métodos SetFloatValue, etc. 

No caso de a primeira letra da cadeia for um ' B', a cadeia se torna um valor binário para ser convertido em uma valor decimal, e usa o método SetOrdValue para armazenar o valor novamente na propriedade. 

Uma vez implementado estes dois métodos (GetValue e SetValue) nós já temos nosso editor de propriedades acabado; nós só temos que registra-lo na VCL. 

<Figura: Bverde.gif - .325 K>Inscrição de editor de propriedades 

De mesmo modo que nós devemos registrar no VCL os componentes, nós temos que registrar os editores de propriedades tambem. Para isto temos o método RegisterPropertyEditor (definida na unidade DsgnIntf): 

procedure RegisterPropertyEditor(PropertyType : PTypeInfo; ComponentClass : TClass;

                                                     const PropertyName : string; EditorClass : TPropertyEditorClass);

PropertyType faz referência ao tipo da propriedade para a qual o editor de propriedades será aplicado. Para dar um valor a este parâmetro nós normalmente usaremos a função TypeInfo, por exemplo, TypeInfo(integer). 

ComponentClass permite restringir o uso do editor de propriedades para a classe específica. Um valor nulo registra o editor para todos os componentes. 

PropertyName especifica o nome da propriedade. Um valor diferente de nulo só registra o editor para a propriedade especificada, enquanto um valor ' ' registra isto para todas as propriedades 

EditorClass especifica o editor de propriedades que registram (a classe). 

Usando estes parâmetros, nós temos a nossa disposição um leque largo de possibilidades para registrar nosso editor de propriedades. Abaixo temos alguns exemplos com nosso editor recentemente criado: 

•RegisterPropertyEditor(TypeInfo(integer), nil, '', TBinIntegerProperty) o editor de registros de propriedades para todos os componentes que têm uma propriedade de tipo inteiro. E é a forma mais global de registrar um componente e afeta a todos os componentes registrados no VCL. Se nós registramos nosso editor deste modo, nós veremos que todas as propriedades do tipo inteiro aparecem em binário! Nós também podemos introduzir um valor novo em decimal ou em binário (colocando a letra B). Nós estamos substituindo o editor de propriedades que Delphi usa para o nosso! :) Esta é a forma mais global de registrar editor de propriedades. •RegisterPropertyEditor(TypeInfo(integer),TMiComponente,'PropriedadeBinaria',TBinIntegerProperty) o editor só registra e exclusivamente para a propriedade ' PropriedadeBinaria' do componente ' TMiComponente'. Esta é a forma restringida de registrar editor de propriedades. 

Eu o aconselho que você registre o editor da forma mais  global possivel para  você experimentar-lo. Então, uma vez você viu todas as propriedades... você o desinstala. 

Você também pode se criar um " componente de mentirinha " e registrar só o editor para o mesmo. Algo assim: 

...

Type

  TMiComponente = class(TComponent)

   ...

     property PropriedadeBinaria : integer read FPropBin write FPropBin;

  ...

  end;

...

<Figura: Bverde.gif - .325 K>Localização de editor de propriedades 

Como nós já mencionamos quando mostramos os passos que deveriam ser continuados criando editor de propriedades, o primeiro deles é criar uma unidade onde localizar o editor. Naquele momento dissemos nós que não uma escolha tão trivial como pudesse parecer em um pricincipio. Naquela unidade deveria se localizar o editor? na mesma unidade onde o componente que tem a propriedade que será publicada ?, em uma unidade separada? Nós temos três possíveis localizações: 

•A primeira opção é localizar o editor de propriedades na mesma unidade em que o componente. Esta é a opção mais intuitiva; porém não é o mais aconselhável, principalmente se o editor usa uma caixa diálogo. 

A razão é que o  Delphi  só usa o editor de propriedades em design - time. De fato, a forma que contém a caixa de diálogo não é conectada com a aplicação, nem o editor de classe de propriedades. Porém, se os recursos da caixa de dialogo são a unica coisa associados a unidade do componente então só estariamos aumentando o tamanho do executável. Então, nós estariamos aumentando o tamanho do executável para nada : (por outro lado, se o editor de propriedades não usa caixas de dialogo, continua ser aconselhavel. 

A segunda opção é localizar o editor de propriedades (se usa uma forma ou uma caixa de diálogo) na mesma unidade da caixa diálogo. Deste modo, a aplicação que usa o componente não conecta ao editor de propriedades seus recursos associados. 

A terceira opção é localizar o editor de propriedades em uma unidade de registro. Uma unidade de registro é uma unidade normal e corrente que contém vários registros condenados correspondendo a componentes diferentes e editores de propriedades que residem em unidades diferentes. Esta é a opção mais aconselhável se o editor de propriedades é usado através de vários componentes.

Então, as últimas duas opções são dependendo do caso, as mais aconselháveis. 

Nas próximas unidades nós veremos exemplos de localizações diferentes dos editores de propriedades que nós iremos construir. 

<Figura: Bverde.gif - .325 K>Codigo Fonte do editor de propriedades. 

unit BinaryPropEd;

interface

uses DsgnIntf;

type

  TBinIntegerProperty = class(TIntegerProperty)

  public

    function GetValue : string; override;

    procedure SetValue(const Value : String); override;

  end;

procedure Register;

implementation

Const

  Bits16 : Array [1..16] of Integer = (32768,16384,8192,4096,2048,1024,512,256,128,64,32,16,8,4,2,1);

function TBinIntegerProperty.GetValue : string;

Var

  Num, i : integer;

begin

  Num:=GetOrdValue;

  Result := '0000000000000000';

  for i := 1 to 16 Do

    if ( Num >= Bits16[i] ) Then

    begin

      Num := Num - Bits16[i];

      Result[i] := '1';

    end;

  if ( Num > 0 ) Then

    raise EPropertyError.Create('Error converting '+IntToStr(GetOrdValue) + ' to binary');

  Insert('B',Result,1);  

end;

procedure TBinIntegerProperty.SetValue(const Value : String);

Var

  i, Total, Longitud : integer;

  NumText : string;

begin

  if UpperCase(Value[1])='B' then

  begin

    NumText:=Copy(Trim(Value),2,Length(Trim(Value))-1);

    NumText:=Copy('0000000000000000',1,16-Length(NumText)) + NumText;

    Total:=0;

    for i:=1 to Length(NumText) do

    begin

      if not (NumText[i] in ['0','1']) then

        raise EPropertyError.Create(NumText[i] + ' is not a valid binary digit')

      else if NumText[i]='1' then

        Total:=Total+Bits16[i];

    end;

    SetOrdValue(Total);

  end

  else

    SetOrdValue(StrToInt(Value));

end;

procedure Register;

begin

  RegisterPropertyEditor(TypeInfo(Integer),nil,'',TBinIntegerProperty);

end;

end.

<Figura: Bverde.gif - .325 K>Introdução 

Na unidade anterior nós aprendemos a operação básica de editor de propriedades e nós desenvolvemos um exemplo de editor de propriedades que trabalhava no Object Inspector (BinaryPropEd). 

Nesta unidade nós desenvolveremos quatro editores de propriedades e um componente que será responsável por eles. 

•TArqProperty é  um editor de propriedades do tipo diálogo. Em alguma ocasião nós desenvolveremos um componente em qual das propriedades deles deveria armazenar o nome de um arquivo. Nós obrigaremos o usuário sofrer fazendo uma rotina para escrever um nome de arquivo correspondente? Claro que não! TArqProperty mostrará para o usuário o diálogo típico OpenDialog  de forma que por meio do editor de propriedades que simplificará o processo de pegar o nome de um arquivo: Não esqueça que a principal tarefa de um editor de propriedades é facilitar a vida do programador. 

<Figura>•TAliasProperty é editor de propriedades inteligente de valores da propriedade identicos ao valor da propriedade DatabaseName do componente TTable: permite a escolha de um alias de banco de dados por meio de uma lista drop-down no próprio Object Inspector. •<Figura>•Aprofundando um pouco mais no tópico dos editores de propriedades tipo caixas de diálogo  nós criaremos um partindo do zero, quer dizer, sem usar uma caixa de diálogo predefinida  como no caso do TArqProperty publicado. Concretamente, nós criaremos um editor que nos permite a introdução de contra-senhas. Imagine isso em uma propriedade de um componente que nós precisamos manter um valor de string do tipo (password). Se nós não criássemos qualquer editor de propriedades, o usuário do componente escreverá o valor diretamente no Object Inspector , mas o que se escreve será completamente visível (não aparecerão como acontece com o componente TEdit). Não muito elegante, verdade? Assim nós criaremos uma caixa de diálogo que nos permitirá escrever e verificar a contra-senha para validar à propriedade. Este será nosso TPasswordProperty . •<Figura>•Por ultimo criaremos um editor de propriedades, TDateTimeProperty,para a introdução de datas. Deste modo nós poderemos introduzir datas em uma propriedade de tipo TDateTime (nós nos lembramos que as datas são do tipo Ponto Flutuante) em vez de ter que introduzir o número equivalente. •<Figura>

Parece interessante, verdade? Mãos para o trabalho:) 

<Figura: Bverde.gif - .325 K>Um componente de teste 

Antes de começar a desenvolver os editores de propriedades nós criaremos um componente que nos permitirá os provar de acordo como quisermos. Este componente se chamara TPrueba e seu código é: 

unit Unidad9;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  DsgnIntf, DB, PasswordForm;

type

  TPrueba = class(TComponent)

  private

    FArquivo : string;

    FAlias : string;

    FData : TDateTime;

    FPassword : string;

  protected

  public

    constructor Create(AOwner : TComponent); override;

  published

    property Arquivo : string read FFichero write FFichero;

    property Alias : string read FAlias write FAlias;

    property Data : TDateTime read FFecha write FFecha;

    property Password : string read FPassword write FPassword;

  end;

...

implementation

...

constructor TPrueba.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  FDAta:=Now;

end;

...

Qualquer coisa do outro mundo. Quatro propriedades de cada um dos editores de propriedades para testarr e um contrutor comum para colocar o valor padrão de data. 

Serve como aviso que pela simplicidade  do componente nós desenvolveremos o componente e os quatro editores de propriedades em uma única unidade (há uma unidade adicional necessária para o TPasswordProperty). Como eu digo, nós faremos isto desta forma pela simplicidade, mas não é a forma mais correta. Em geral, cada editor de propriedades deveria entrar em uma unidade independente do próprio componente, como nós já vimos na unidade 8. 

<Figura: Bverde.gif - .325 K>O editor de propriedades TArqProperty 

Eu fixei na propriedade tal Arquivo e como nós implementamos isto no componente TPrueba. Claro que, é do tipo string. Nós também não definimos nenhum métodos Set Get, pois o read/write de valores nesta propriedade é feito diretamente no campo associado (FArquivo). Tal como é, quando um usuário do componente quer introduzir um valor nesta propriedade, ele terá que escrever isto à mão, o que pode acabar sendo uma tarefa tediosa se o arquivo em questão esta em um caminho longo. De forma que, por generosidade nossa, nós decidimos o dar uma mão: nós construiremos um editor de propriedades para ele. 

O Delphi já incorporá um componente que administra isso muito bem para a de arquivos: TOpenDialog, de forma que nós veremos como usar isto melhor em nosso editor. 

O primeiro passo é decidir de quem nosso editor se derivará. Em nosso caso, desde que a propriedade arquivo é do tipo string, nós decidimos herdar de TStringProperty. Logo nós deveremos decidir se a propriedade será editavel no próprio editor de objetos ou usará uma caixa de diálogo. Neste caso, sem dúvida nós deveremos escolher a segunda opção. Claro que isto não tira a chance do escrever o valor da propriedade diretamente no Object Inspector. 

Agora então, como indicar ao Delphi que se trata de um editor de propriedades do tipo caixa de dialogos? Para  isto, nós devemos usar outro método da classe base de todos os editores de propriedades (TPropertyEditor): o método GetAttributes Este método é uma função que determina que características terá o editor de propriedades. As características queridas deveriam os devolver como resultado desta função; este resultado é do tipo TPropertyAttributes O tipo TPropertyAttributes é um grupo (Set) isso pode levar os valores seguintes: 

  

Atributo Descrição paValueList Especifica que o editor devera mostrar uma lista com os possíveis valores pela propriedade. Para encher a lista o método GetValues é usado paSortList Só válido se paValueList é selecionado. Especifica que a lista de valores será mostrada ordenada. paSubProperties Indica que o editor define subpropiedades para mostrar à direita (p.e. a propridade fonte do TForm usa isto). Para gerar a lista de propriedades o método GetProperties é usado. paDialog Indica que o editor de propriedades devera mostrar uma caixa de diálogo em resposta para ao método Edit. (p.e. a propriedade glyph de um TSpeddButton). Deste modo quando selecionando a propriedade aparecerá um botão com '... '. paMultiSelect Possibilita a multi-seleção de opçõespaAutoUpdate O Object Inspecto o método SetValue toda vez que o valor da propriedade é mudado. paReadOnly O usuário não pode modificar o valor da propriedade paRevertable Especifica se a propriedade pode recuperar seu valor original 

Devido a este quadro, é fácil de implementar o método GetAtributes em nosso editor de propriedades: 

  interface 

  ... 

  TArqProperty = class(TStringProperty) 

  public 

    function GetAttributes : TPropertyAttributes; override; 

    procedure Edit; override; 

  end; 

  ... 

function TArqProperty.GetAttributes : TPropertyAttributes; 

begin 

  Result:=[paDialog]; 

end; 

Novamente por causa da simplicidade nós só ativamos o atributo paDialog, é claro que nós poderíamos ativar outro atributo como paMultiSelect, etc. Neste caso, escreveriamos [paDialog, paMultiSelect] 

Nós só saberemos que o Delphi ativará a caixa de diálogo, quando o usuário clica no botão '... ' ou da um duplo clique na propriedade,o Delphi invocará o método Edit de nosso editor de propriedades. Neste método nós deveremos colocar o codigo necessario para mostrar a caixa de dialogo de abertura de arquivos e colocar o nome do arquivo na propriedade se o usuário clicar no OK.. Este método que nós devemos reimplementar (overridee, ver a seção de interface), é deste modo: 

...

procedure TArqProperty.Edit;

var

  OpenDialog : TOpenDialog;   {TOpenDialog está na unidade Dialogs, que esta na clausula uses}

begin

  OpenDialog:=TOpenDialog.Create(Application);   {Criamos a caixa de diálogo}

  try

    OpenDialog.Filter:='All files|*.*';    {Colocamos suas propriedades iniciais}

    if OpenDialog.Execute then   {Se o usuário clica OK...}

      SetStrValue(OpenDialog.FileName);   {Colocamos o novo valor na propriedade}

  finally

    OpenDialog.Free;   {Liberamos a caixa de diálogo}

  end;

end;

...

Mais fácil impossível. Só temos que comentar que para termos certeza da liberação de recursos (neste caso da caixa de diálogo) usamos o bloco try... finally . 

É só! Com 10 ou 15 linhas de código nós aliviamos o pobre usuário de ficar sofrendo com isso tudo. Nós agora registraremos o editor: 

procedure Register;

begin

   ...

  RegisterPropertyEditor(TypeInfo(string),TPrueba,'Arquivo',TArqProperty);

  ...

end;

<Figura: Bverde.gif - .325 K>O editor de propriedades TAliasProperty 

Vamos contruir agora um editor de propriedades para a propriedade Alias. Como nós já sabemos, o componente TTable tem uma propriedade denominada DatabaseName que especifica o alias que conecta ao banco de dados. Selecionar um valor por esta propriedade é muito facil, basta escolhermos um na lista. Este é o comportamento que nós queremos para a nossa propriedade Alias. Nós poderíamos procurar no codigo fonte do VCL e tentar também registrar o editor de propriedades da propriedade DatabaseName de forma que isto incluiria nossa nova propriedade, mas como a construção de um editor deste tipo é muito simples, nós mesmos desenvolveremos isto. 

A seção de interface de nosso editor é o seguinte: 

...

TAliasProperty = class (TStringProperty)

  public

    function GetAttributes : TPropertyAttributes; override;

    procedure GetValues(Proc : TGetStrProc); override;

  end;

...

O método GetAttributes conhecido na seção anterior. Basta colocar o atributo paValueList (indica a lista de opções) e o paSortList (indica que as opções estarão ordenadas alfabeticamente). 

function TAliasProperty.GetAttributes : TPropertyAttributes;

begin

    Result:=[paValueList, paSortList];

end;

Agora nós preencheremos a lista de aliases. Para isto, reimplementaremos (override) no método GetValues. Este método recebe um único parâmetro: um ponteiro para método. Realmente este ponteiro realmente da referência ao metodo interno Add, que adiciona uma string na lista. Os diversos elementos inseridos na lista pelo método GetValues que invoca o método refereciado pelo ponteiro  e  transforma isto em um valor do tipo string. Soa complicado, não? Não se preocupe com isso, pois não é tão dificil; so significa que temos que usar a senteça Proc(string) para cada elemento a adicionar à lista. Em nosso caso, como nós queremos adicionar os nomes dos aliases existentes, nós faremos uma ligação a Proc(nome do alias) e adiciona todos os valores. Previamente, nós teremos obtido esses aliases existente por meio do método GetAliasList do objeto TSession: 

procedure TAliasProperty.GetValues(Proc : TGetStrProc);

Var

  AliasList : TStringList;   {lista com os alias existentes}

  i : integer;

begin

  try

    AliasList := TStringList.Create;   {Criamos a lista}

    Session.GetAliasNames(AliasList);   {Obtemos os alias existentes}

    for i:=0 to AliasList.Count - 1 do     {Para cada alias...}

      Proc(AliasList[i]);   {...fazemos a chamada ao método Proc}

  finally

    AliasList.Free;   {Liberamos a lista}      

  end;

end;

Com isto nós já construímos já nosso novo editor de propriedades. Só falta registrá-lo: 

procedure Register;

begin

   ...

  RegisterPropertyEditor(TypeInfo(String),TPrueba,'Alias',TAliasProperty);

  ...

end;

<Figura: Bverde.gif - .325 K>O editor de propriedades TPasswordProperty 

Vamos construir um outro editor de propriedades de caixa de dialogos. A operação dela será semelhante ao TArqProperty. Nós colocaremos paDialog para indicar que é editor de caixa de diálogo e no método Edit estara todo o codigo do componente. 

A diferença principal é que nosso componente não usará uma caixa de dialogo existente, nós criaremos tambem uma caixa de dialogos, com dois labels e dois edit boxes: 

<Figura>

A coisa mais importante é nomear à propriedade PasswordChar do dois TEdit, para o caráctere ' * ' de forma que este caráter aparece quando um usuário digita uma contra-senha. Também, no evento OnCloseQuery conferirá a validade da contra-senha introduzida. 

Os Edits foram nomeados de PW1 e PW2 

O código da unidade PasswordForm: 

unit PasswordForm;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  StdCtrls, Buttons;

type

  TfrmPassword = class(TForm)

    lpwd: TLabel;

    lVpwd: TLabel;

    PW1: TEdit;

    PW2: TEdit;

    bOK: TBitBtn;

    bCancel: TBitBtn;

    procedure FormCloseQuery(Sender: TObject; var CanClose: Boolean);

  private

  public

  end;

var

  frmPassword: TfrmPassword;

implementation

{$R *.DFM}

procedure TfrmPassword.FormCloseQuery(Sender: TObject;

  var CanClose: Boolean);

begin

  if (ModalResult=mrOK) then

    if (PW1.Text = '') then

    begin

      ShowMessage('Deve se inserir a contra-senha');

      CanClose:=False;

     end

  else if (ModalResult=mrOK) and (PW1.Text <> PW2.Text) then

   begin

    ShowMessage('Verificação inválida. Por favor retente');

    CanClose:=False;

   end;

end;

end.

Ya hemos construido el cuadro de diálogo, ahora sólo nos resta "engancharlo" al editor de

propiedades. Para ello, efectuamos la llamada al form en el método Edit del editor de propiedades.

Veamos como queda el código del editor:

function TPasswordProperty.GetAttributes : TPropertyAttributes;

begin

  Result:=[paDialog];

end;

function TPasswordProperty.GetValue : string;

begin

  Result:=Format('(%s)',[GetPropType^.Name]);

end;

procedure TPasswordProperty.Edit;

begin

  frmPassword := TfrmPassword.Create(Application);

  try

    frmPassword.Caption:=GetComponent(0).Owner.Name+'.'+

      GetComponent(0).Name+'.'+GetName+' - '+

      frmPassword.Caption;

    frmPassword.PW1.Text:=GetStrValue;

    frmPassword.PW2.Text:=frmPassword.PW1.Text;

    if frmPassword.ShowModal = mrOK then

      SetStrValue(frmPassword.PW1.Text)

  finally

    frmPassword.Free;

  end;

end;

Nós já construímos a caixa de diálogo, agora só nos falta encaixa-lo no editor de propriedades. Para isto, nós fazemos a ligação à forma com o método Edit do editor de propriedades. Vamos ver como é o codigo: 

function TPasswordProperty.GetAttributes : TPropertyAttributes;

begin

  Result:=[paDialog];

end;

function TPasswordProperty.GetValue : string;

begin

  Result:=Format('(%s)',[GetPropType^.Name]);

end;

procedure TPasswordProperty.Edit;

begin

  frmPassword := TfrmPassword.Create(Application);

  try

    frmPassword.Caption:=GetComponent(0).Owner.Name+'.'+

      GetComponent(0).Name+'.'+GetName+' - '+

      frmPassword.Caption;

    frmPassword.PW1.Text:=GetStrValue;

    frmPassword.PW2.Text:=frmPassword.PW1.Text;

    if frmPassword.ShowModal = mrOK then

      SetStrValue(frmPassword.PW1.Text)

  finally

    frmPassword.Free;

  end;

end;

Só há uma coisa nova: no método GetValue nós não queremos que seja mostrado

o valor da contra-senha, então nós poderíamos mostrar outra coisa. Poderia

ser uma cadeia de asteriscos...    

procedure Register;

begin

  ...

  RegisterPropertyEditor(TypeInfo(String),TPrueba,'Password',TPasswordProperty);

end;

<Figura: Bverde.gif - .325 K>O editor de propriedades TDateTimeProperty 

Para terminar esta unidade nós desenvolveremos um editor de tipo de propriedades TDateTime. 

Como nós já sabemos o tipo TDateTime do Delphi é do tipo ponto flutuante, de forma que se nós temos em um componente com uma propriedade de tipo DateTime, e nós não registramos qualquer editor de propriedades para esta propriedade, o usuário terá que escrever a data querida em formato decimal. Que ruim! Principalmente se nós pensamos que com seis linhas de código é resolvido o problema: 

function TDateTimeProperty.GetValue : string;

begin

  Result:=DateTimeToStr(GetFloatValue);

end;

procedure TDateTimeProperty.SetValue(const Value : string);

begin

  SetFloatValue(StrToDateTime(Value));

end;

procedure Register;

begin

  ...

  RegisterPropertyEditor(TypeInfo(TDateTime),TPrueba,'Fecha',TDateTimeProperty);

end;

Nada novo até aqui. Nós só nos limitamos para chamar os métodos GetFloatValue e SetFloatValue. 

<Figura: Bverde.gif - .325 K>Conclusões 

Com isto que nós terminamos o tópico dos editores de propriedades. Nós aprendemos a criar e  publicar editores editaveis no Object Inspector, que pode ser preenchido diretamente na propriedade ou por uma caixa de dialogo. O poder de editor de propriedades é imenso: por meio do métodos SetValue e GetValue podemos fazer uma verificação do valor da propriedade e agir depois. Também, nosso editor pode fazer mil coisas: consultar um arquivo ini, fazer cálculos complexos ou até mesmo operar com um banco de dados! 

Mas os editores de propriedades têm duas limitações: 

•Eles só operam em design-time, aspecto que nós deveríamos nos lembrar que tem coisas que o usuário pode querer fazer em run-time. Um exemplo claro é as propriedades de tipo Lista p.e. os artigos de propriedade de um TListBox. Em design-tmie o editor de propriedades especifico permite o usuário de um modo intuitivo adicionar e eliminar elementos. Em troca, em run-time, ele tem os métodos Add e Delete para operar com a lista. •

Editor de propriedades só opera com o valor de uma propriedade. Quer dizer, nós não podemos alterar o valor de propriedades diferentes ao mesmo tempo, desde os métodos GetValue e SetValue fazem  referência à propriedade que está publicada, e não para o resto deles. Então, quando nós queremos alterar várias propriedades de um componente ao mesmo tempo, nós usaremos Editor de componentes, tópico isto que será objeto de nossa próxima unidade. 

<Figura: Bverde.gif - .325 K>Codigo fonte dos editores de propriedades.

 unit Unidad9;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  DsgnIntf, DB, PasswordForm;

type

  TPrueba = class(TComponent)

  private

    FArquivo : string;

    FAlias : string;

    FData : TDateTime;

    FPassword : string;

  protected

  public

    constructor Create(AOwner : TComponent); override;

  published

    property Arquivo : string read FFichero write FFichero;

    property Alias : string read FAlias write FAlias;

    property Data : TDateTime read FFecha write FFecha;

    property Password : string read FPassword write FPassword;

  end;

  TArqProperty = class(TStringProperty)

  public

    function GetAttributes : TPropertyAttributes; override;

    procedure Edit; override;

  end;

  TAliasProperty = class (TStringProperty)

  public

    function GetAttributes : TPropertyAttributes; override;

    procedure GetValues(Proc : TGetStrProc); override;

  end;

  TDateTimeProperty = class(TFloatProperty)

    function GetValue : string; override;

    procedure SetValue(const Value  : string); override;

  end;

  TPasswordProperty = class(TPropertyEditor)

    function GetAttributes : TPropertyAttributes; override;

    function GetValue : string; override;

    procedure Edit; override;

  end;

procedure Register;

implementation

constructor TPrueba.Create(AOwner : TComponent);

begin

  inherited Create(AOwner);

  FFecha:=Now;

end;

function TFicheroProperty.GetAttributes : TPropertyAttributes;

begin

  Result:=[paDialog];

end;

procedure TArqProperty.Edit;

var

  OpenDialog : TOpenDialog;

begin

  OpenDialog:=TOpenDialog.Create(Application);

  try

    OpenDialog.Filter:='All files|*.*';

    if OpenDialog.Execute then

      SetStrValue(OpenDialog.FileName);

  finally

    OpenDialog.Free;

  end;

end;

function TAliasProperty.GetAttributes : TPropertyAttributes;

begin

    Result:=[paValueList, paSortList];

end;

procedure TAliasProperty.GetValues(Proc : TGetStrProc);

Var

  AliasList : TStringList;

  i : integer;

begin

  try

    AliasList := TStringList.Create;

    Session.GetAliasNames(AliasList);

    for i:=0 to AliasList.Count - 1 do

      Proc(AliasList[i]);

  finally

    AliasList.Free;

  end;

end;

function TDateTimeProperty.GetValue : string;

begin

  Result:=DateTimeToStr(GetFloatValue);

end;

procedure TDateTimeProperty.SetValue(const Value : string);

begin

  SetFloatValue(StrToDateTime(Value));

end;

function TPasswordProperty.GetAttributes : TPropertyAttributes;

begin

  Result:=[paDialog];

end;

function TPasswordProperty.GetValue : string;

begin

  Result:=Format('(%s)',[GetPropType^.Name]);

end;

procedure TPasswordProperty.Edit;

begin

  frmPassword := TfrmPassword.Create(Application);

  try

    frmPassword.Caption:=GetComponent(0).Owner.Name+'.'+

      GetComponent(0).Name+'.'+GetName+' - '+

      frmPassword.Caption;

    frmPassword.PW1.Text:=GetStrValue;

    frmPassword.PW2.Text:=frmPassword.PW1.Text;

    if frmPassword.ShowModal = mrOK then

      SetStrValue(frmPassword.PW1.Text)

  finally

    frmPassword.Free;

  end;

end;

procedure Register;

begin

  RegisterComponents('Pruebas', [TPrueba]);

  RegisterPropertyEditor(TypeInfo(string),TPrueba,'Arquivo,TArqProperty);

  RegisterPropertyEditor(TypeInfo(String),TPrueba,'Alias',TAliasProperty);

  RegisterPropertyEditor(TypeInfo(TDateTime),TPrueba,'Data',TDateTimeProperty);

  RegisterPropertyEditor(TypeInfo(String),TPrueba,'Password',TPasswordProperty);

end;

end.

unit PasswordForm;

interface

uses

  Windows, Messages, SysUtils, Classes, Graphics, Controls, Forms, Dialogs,

  StdCtrls, Buttons;

type

  TfrmPassword = class(TForm)

    lpwd: TLabel;

    lVpwd: TLabel;

    PW1: TEdit;

    PW2: TEdit;

    bOK: TBitBtn;

    bCancel: TBitBtn;

    procedure FormCloseQuery(Sender: TObject; var CanClose: Boolean);

  private

  public

  end;

var

  frmPassword: TfrmPassword;

implementation

{$R *.DFM}

procedure TfrmPassword.FormCloseQuery(Sender: TObject;

  var CanClose: Boolean);

begin

  if (ModalResult=mrOK) then

    if (PW1.Text = '') then

    begin

      ShowMessage('Debe introducir una contraseña');

      CanClose:=False;

     end

  else if (ModalResult=mrOK) and (PW1.Text <> PW2.Text) then

   begin

    ShowMessage('Verificación fallida. Por favor reintente');

    CanClose:=False;

   end;

end;

end.

